Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Surveying



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAMPLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE"	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

* Requires one memory module.

** Requires a card reader and one memory module.

SPIRAL CURVE LAYOUT

(This program requires one memory module)

This program calculates the field data for layout of spiral curves by either of two methods; 1) deflection angles and chord distances from the beginning station of the spiral (PS), or 2) tangent distances and offsets to the spiral. The required information on the spiral is the PS station, radius (or degree of curve) of the central circular curve, and length of the spiral. For equal exit and entrance spirals to a central circular curve, the station (PI) and the angle (TΔ) at the intersection of the tangents is required.

Field data for any specified station can be calculated, or if a stationing interval is given, the field data for successive stations can be computed automatically.



L = Length of spiral curve from PS to PSC.

D = Degree of central circular curve = central angle per 100 ft. of arc.



ℓ_s = Length of spiral curve from PS to STA.

Equations:

Reference: "Standard Highway Spiral," Oregon State Highway Division Technical Bulletin No. 20-Revised, Oregon Department of Transportation, August 1973, by H.W. Libby and E.M. Booth.

$$R = \frac{18{,}000}{\pi D}$$

$$U = Y_s - X_s \cot S$$

$$S = \frac{90L}{\pi R}$$

$$p = X_s - R(1 - \cos S)$$

$$\theta_s = \frac{S\pi}{180} = \frac{L}{2R}$$

$$q = Y_s - R \sin S$$

$$\theta = \left(\frac{\ell_s}{L}\right)^2 \left(\frac{\pi S}{180}\right)$$

$$T_s = q + (R + p) \tan \frac{T_\Delta}{2}$$

$$X = TO = \ell_s \left[\frac{\theta}{3} - \frac{\theta^3}{42} + \frac{\theta^5}{1320} - \frac{\theta^7}{75{,}600} + \frac{\theta^9}{6{,}894{,}720}\right]$$

$$Y = TD = \ell_s \left[1 - \frac{\theta^2}{10} + \frac{\theta^4}{216} - \frac{\theta^6}{9360} + \frac{\theta^8}{685{,}440}\right]$$

$$CD = \sqrt{X^2 + Y^2}$$

$$ANG = \tan^{-1} \frac{X}{Y}$$

where:

R = Radius of central curve

D = Degree of curve of central curve

S = Central angle of each spiral

L = Length of each spiral (distance along curve)

θ_s = S expressed in radians

θ = Central angle of spiral curve section between PS and station on spiral (between beginning station of spiral and any other station on spiral)

ℓ_s = Length of spiral, measured along curve, between PS and any station on spiral

X = TO = Tangent offset = Perpendicular from tangent to station on curve, where tangent is line which is tangent to curve at PS

Y = TD = Tangent distance = distance along tangent from PS to right angle intersection of tangent and tangent offset X

U = Distance along tangent from PS to PSI where PSI is intersection of spiral tangents; i.e., intersection of tangents from PS and PSC

Y_s = Y for station at PSC

X_s = X for station at PSC

p = The offset from the PS tangent to the point where the tangent to the circular curve (extended backward or forward) becomes parallel to the main tangent

q = The distance along the tangent from the PS to the offset p.

T_S = Distance along tangent from PS to PI, where PI is the intersection of the main tangents from PS and PT

T_Δ = Total angle, equal to $\Delta + 2S$, the central angle of the circular curve plus the central angles of both equal spirals

CD = Length of chord from PS to any station on spiral curve

ANG = Deflection angle from main tangent to chord CD

Example:

Compute field data for layout of entrance and exit transition spirals 300 feet long. The PI station is 35 + 290.48 and the intersection angle is 40°00'00". The spirals are transitions to a central circular curve (degree of curve = 6°00'). Deflection angles and distances are needed at 100 foot intervals from the first even station inside the PS (PT).

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
XEQ [ALPHA] SPIRAL [ALPHA]	PS OR PI, ↑, ∠
35290.48 [ENTER ↑] 40 [F]	D OR R
6 [B]	L=?
300 [C]	SPIRAL DATA
[R/S]*	L=300.0000
[R/S]*	S=9.0000
[R/S]*	D=5.5960
[R/S]*	R=954.9297
[R/S]*	PT=35,758.2759
[R/S]*	PI=35,290.4800
[R/S]*	ENTR SPIRAL
[R/S]*	PSC=35,091.6093
[R/S]*	SPI=34,991.8684
[R/S]*	PS=34,791.6093
[R/S]*	STA=?
34800 [D]	STA=34,800.0000
[R/S]*	CD=8.3907
[R/S]*	∠ =0.0008
[R/S]*	STA, IV OR EX
100 [H]	PS OR TN

Keystrokes:	Display:
[D]	STA=34,900.0000
[R/S]*	CD=108.3887
[R/S]*	∆ =0.2330
[R/S]*	STA=35,000.0000
[R/S]*	CD=208.3375
[R/S]*	∆ =1.2651
[R/S]*	STA=35,091.6093
[R/S]*	CD=299.6711
[R/S]*	∆ =2.5958
[R/S]*	STA, IV OR EX
[E]	EXIT SPIRAL
[R/S]*	PCS=35,458.2759
[R/S]*	SPI=35,558.0168
[R/S]*	PT=35,758.2759
[R/S]*	STA=?
35700 [D]	STA=35,700.0000
[R/S]*	CD=58.2758
[R/S]*	∆ =0.0648
[R/S]*	STA, IV OR EX
100 [CHS] [H]	PS OR TN
[D]	STA=35,600.0000
[R/S]*	CD=158.2625
[R/S]*	∆ =0.5006
[R/S]*	STA=35,500.0000
[R/S]*	CD=258.1204
[R/S]*	∆ =2.1324
[R/S]*	STA=35,458.2759
[R/S]*	CD=299.6711
[R/S]*	∆ =2.5958
[R/S]*	STA, IV OR EX

*Press [R/S] if printer is not attached.

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize program		[XEQ] SPIRAL	PS OR PI,↑,Δ
3.	Input station at beginning of spiral,	PS	[A]	D OR R
	or station	PI	[ENTER↑]	(PI)
	and angle at intersection of tangents	T (DMS)	[F]	D OR R
4.	Input degree of curve of circular curve,	D (DMS)	[B]	L=?
	or radius of circular curve.	R	[G]	L=?
5.	Input length of spiral curve.	L	[C]	SPIRAL DATA
			[R/S]*	L=( )
			[R/S]*	S=( )
			[R/S]*	R=( )
6.	If PS was input go to step 8.			
7.	Additional output if PI was input		[R/S]*	PT=( )
			[R/S]*	PI=( )
8.	For additional output for entrance spiral		[R/S]*	ENTR SPIRAL
			[R/S]*	PSC=( )
			[R/S]*	SPI=( )
			[R/S]*	PS=( )
			[R/S]*	STA=?
9.	Input station	STA		
10.	Compute:			
a.	PS deflection		[D]	STA=( )
			[R/S]*	CD=( )
			[R/S]*	Δ=( )
	See prompt for Station Interval or Exit Spiral		[R/S]*	STA, IV OR EX
b.	or tangent offset		[I]	STA=( )
			[R/S]*	TO=( )

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
			[R/S]*	TD=( )
	See prompt for Station, Interval, or Exit Spiral		[R/S]*	STA, IV OR EX
11.	Repeat steps 9 and 10 for next station or			
	go to step 12 for automatic stationing or			
	go to step 13 for exit spiral			
12.	Input stationing interval** (IV)	IV	[H]	PS OR IV
	Go to step 10 and select layout method.			
	Program will begin computing field data			
	from current station.			
13.	If PI was input, field data for exit spiral			
	(EX) can be computed.		[E]	EXIT SPIRAL
			[R/S]*	PCS=( )
			[R/S]*	SPI=( )
			[R/S]*	PT=( )
			[R/S]*	STA=?
	Go to step 9 if more field data is desired			
14.	To restart program with different inputs			
	go to step 3.		[J]	PS OR PI,↑,Δ
**	Note: it is not necessary to input the			
	interval if it has been done previously.			
	Just press the [H] key without keying in a			
	number.			
*	Press [R/S] if printer is not attached.			

Program Listings

```
 01◆LBL "SPI
RAL"
 02 CF 02          Initialize and
 03 SF 21          load constants
 04 SF 27
 05 DEG
 06 FIX 4
 07 CF 01
 08 3
 09 STO 11
 10 -10
 11 STO 12
 12 -42
 13 STO 13
 14 216
 15 STO 14
 16 1320
 17 STO 15
 18 -9360
 19 STO 16
 20 -75600
 21 STO 17
 22 685440
 23 STO 18
 24 6894720
 25 STO 19
 26◆LBL J
 27 CF 22
 28 "PS OR P
I, ↑, ∠"
 29 PROMPT
 30◆LBL A          PS was input
 31 STO 01
 32 STO 02
 33 SF 01
 34 GTO 43
 35◆LBL F          PI was input
 36 HR
 37 2
 38 /
 39 STO 05
 40 X<>Y
 41 STO 06
 42 CF 01
 43◆LBL 43
 44 CF 04
 45 "D OR R"
 46 PROMPT
 47◆LBL B          D was input
 48 HR
 49 1/X
 50 R-D
 51 1 E2
 52 *
 53◆LBL G          R
 54 CF 04
 55 STO 20
 56 "L=?"
 57 PROMPT
 58◆LBL C          Ls input
 59 STO 22
 60 ENTER↑
 61 ST+ 02
 62 RCL 20
 63 /
 64 2
 65 /
 66 STO 00
 67 R-D
 68 STO 23
 69 HMS
 70 1 E2
 71 RCL 20
 72 /
 73 R-D
 74 HMS
 75 CLA
 76 XEQ 40         Output spiral
 77 "⊢ DATA"       data
 78 ADV
 79 AVIEW
 80 RCL 20
 81 "L="
 82 ARCL T
 83 AVIEW
 84 "S="
 85 ARCL Z
 86 AVIEW
 87 "D="
 88 ARCL Y
 89 AVIEW
 90 "R="
 91 ARCL X
 92 AVIEW
 93 CLD
 94 ADV
 95 FS? 01         Was PS input?
 96 GTO 15
 97 0
 98 STO 01
 99 RCL 22
100 STO 02
```

Program Listings

```
101 XEQ 09
102 RCL 23
103 RCL 20
104 P-R
105 RDN
106 -                    Ys-Rsin(s)
107 RDN
108 +                    Xs+Rcos(s)
109 RCL 05
110 TAN
111 *
112 R↑
113 +
114 STO 24
115 RCL 06
116 X<>Y
117 -
118 STO 01
119 ST+ 02
120 RCL 05
121 RCL 23
122 -
123 2
124 *
125 STO 21
126 D-R
127 RCL 20
128 *
129 RCL 22
130 2
131 *
132 +
133 +
134 STO 07                PT = PX + Lt
135 RCL 06
136 RCL 01
137 0
138 RDN
139 "PT="
140 ARCL Z
141 AVIEW
142 "PI="
143 ARCL Y
144 AVIEW
145 CLD
146*LBL 15                Compute PS, SPI,
147 XEQ 09               PSC
148 X<>Y
149 RCL 23
150 TAN
151 /
152 -
153 FS? 02               Exit spiral?
154 CHS
155 STO 24
156 0
157 RCL 01
158 RCL 22
159 +                    PSC = PS + Ls
160 STO 03
161 RCL 01
162 RCL 24
163 +                    SPI = PS + U
164 RCL 01
165 ADV
166 "ENTR "              Output and prompt
167 FS? 02
168 "EXIT "
169 XEQ 40
170 AVIEW
171 "PSC="
172 FS? 02
173 "PCS="
174 ARCL Z
175 AVIEW
176 "SPI="
177 ARCL Y
178 AVIEW
179 "PS="
180 FS? 02
181 "PT="
182 ARCL X
183 AVIEW
184 ADV
185*LBL 45                Prompt for next
186 "STA"                station
187 FC?C 02
188 GTO 01
189 "⊢=?"
190 GTO 03
191*LBL 01
192 FC? 04
193 "⊢=?"
194 FS? 04
195 "⊢, IV O
R EX"
196*LBL 03
197 PROMPT
198*LBL E                 Exit spiral
199*LBL 18
200 RCL 07
201 RCL 01
```

Program Listings

```
202 STO 07
203 X<>Y
204 STO 01
205 RCL 22
206 CHS
207 STO 22
208 +
209 STO 02
210 SF 02
211 GTO 15
212◆LBL H          Exit spiral
213 CF 00
214 FS?C 22
215 STO 04
216 RCL 04
217 "PS OR T
N"
218 PROMPT
219◆LBL D          DS Def
220 CF 01
221 GTO 19
222◆LBL I          TN off
223 SF 01
224◆LBL 19
225 SF 04          Computing field
226 RCL 02         data
227 -
228 FS?C 22        STA input?
229 GTO 00
230 RCL 04
231 CF 02
232 X<0?           ITV neg?
233 SF 02          Exit spiral
234 ABS
235 RCL 03
236 RCL 02
237 -
238 ABS
239 X<=Y?
240 GTO 00         End of spiral
241 X<>Y
242 GTO 07
243◆LBL 00         Stop
244 SF 00
245◆LBL 07
246 FS?C 02        Exit spiral?
247 CHS
248 ST+ 02
249 RCL 02
250 RCL 01
251 -
252 RCL 22
253 /
254 X↑2
255 RCL 23
256 D-R
257 *
258 STO 00
259 XEQ 09
260 FS? 01
261 GTO 04
262 R-P
263 X<>Y
264 HMS
265◆LBL 04
266 RCL 02
267 0
268 RDN
269 "STA="          Output field data
270 ARCL X
271 AVIEW
272 "CD="
273 FS? 01
274 "TO="
275 ARCL Z
276 AVIEW
277 "∠="
278 FS? 01
279 "TD="
280 ARCL Y
281 AVIEW
282 ADV
283 CLD
284 FS?C 00
285 GTO 45
286 GTO 19
287◆LBL 09          Computes Y & X
288 CF 22
289 9
290 STO 25
291 0
292 STO 08
293 STO 09
294◆LBL 02          Loop point
295 RCL 00
296 RCL 25
297 Y↑X
298 10
299 ST+ 25
300 CLX
301 RDN
302 RCL IND
25
```

Program Listings

```
303 /
304 10
305 ST- 25
306 CLX
307 RDN
308 RCL 08
309 RCL 09
310 STO 08
311 RDN
312 +
313 STO 09
314 DSE 25
315 GTO 02
316 RCL 09
317 RCL 02
318 RCL 01
319 -
320 ABS
321 *
322 LASTX
323 RCL 08
324 1
325 +
326 *
327 RTN
328◆LBL 40
329 "⊢SPIRAL
"
330 RTN
331 END
```

X

Y

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	θs, θ	50	
	PS, PT		
	STA		
	PSC, PCS		
	ITV		
05	TΔ/2	55	
	PI		
	PT, PS		
	Y		
	X		
10		60	
	3		
	-10		
	-42		
	216		
15	1320	65	
	-9360		
	-75600		
	685440		
	6894720		
20	R		
	Δ		
	Ls'-Ls		
	S		
	Ts, U		
25	Used		
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 026 TOT. REG. 116 USER MODE

ENG ____ FIX 4 SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	STOP w/this station	AUTO station
01	C	PS was input	PI was input
02	C	Solving for exit spiral	not exit spiral
04	C	Have solved for field data	
21	C	Printer Enable	Printer Disable
22	C	Numeric Input	
27	C	User On	User Off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input PS	A	Input PI,T ~~A~~	F
Input D	B	Input radius	G
Input L	C	Input interval	H
STA PS, DEF	D	Solve for tang offset	I
STA TN, OFF	E	Restart	J

SPIRAL CURVE LAYOUT

PROGRAM REGISTERS NEEDED: 90

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 – 3)

ROW 2 (3 – 10)

ROW 3 (11 – 16)

ROW 4 (17 – 20)

ROW 5 (20 – 24)

ROW 6 (24 – 28)

ROW 7 (28 – 28)

ROW 8 (29 – 36)

ROW 9 (37 – 45)

ROW 10 (45 – 51)

ROW 11 (51 – 57)

ROW 12 (58 – 66)

ROW 13 (67 – 75)

ROW 14 (76 – 80)

ROW 15 (80 – 86)

ROW 16 (87 – 93)

ROW 17 (94 – 101)

ROW 18 (101 – 111)

ROW 19 (112 – 121)

ROW 20 (122 – 131)

ROW 21 (132 – 140)

ROW 22 (141 – 147)

ROW 23 (147 – 156)

ROW 24 (157 – 166)

ROW 25 (166 – 169)

ROW 26 (169 – 173)

ROW 27 (173 – 178)

ROW 28 (179 – 183)

ROW 29 (184 – 189)

ROW 30 (189 – 194)

ROW 31 (195 – 196)

ROW 32 (197 – 206)

ROW 33 (207 – 213)

ROW 34 (214 – 217)

ROW 35 (218 – 224)

ROW 36 (224 – 232)

ROW 37 (233 - 242)

ROW 38 (243 - 252)

ROW 39 (252 - 260)

ROW 40 (261 - 269)

ROW 41 (269 - 274)

ROW 42 (274 - 279)

ROW 43 (280 - 286)

ROW 44 (287 - 296)

ROW 45 (297 - 305)

ROW 46 (305 - 315)

ROW 47 (316 - 328)

ROW 48 (328 - 331)

TWO INSTRUMENT RADIAL SURVEY

This program uses a two instrument radial survey technique to determine the coordinates of a point. The required information consists of the coordinates of the theodolite, elevation and height of instrument for the theodolite, reference azimuth (or bearing), angle right at the theodolite from reference azimuth to distance meter, slope distance and vertical (or zenith) angle from theodolite to distance meter, angle right at theodolite from reference azimuth to unknown point, slope distance from distance meter to point, vertical (or zenith) angle from theodolite to point and height of instrument for the prism at the point. The horizontal distance and azimuth from the theodolite to the point are computed as well as the coordinates and elevation of the point.

Equations:

$$\cos \alpha = \sin (AR\ DM) \sin (ZA\ DM) \qquad \cos \gamma = \cos (ZA\ DM)$$

$$\cos \beta = \cos (AR\ DM) \sin (ZA\ DM) \qquad SD\ DM = \frac{HD\ DM}{\sin (ZA\ DM)}$$

$$\cos \theta = \cos \alpha \sin (AR\ P) \sin (ZA\ P) + \cos \beta \cos (AR\ P) \sin (ZA\ P) + \cos \gamma \cos (ZA\ P)$$

$$HD\ P = \left[(SD\ DM) \cos \theta + \sqrt{(SD\ P)^2 - (SD\ DM)^2 + (SD\ DM)^2 \cos^2 \theta} \right] \sin (ZA\ P)$$

where:

$\cos \alpha$, $\cos \beta$, $\cos \gamma$ = Direction cosines of line from theodolite to distance meter

AR DM = Angle right from backsight to distance meter

ZA DM = Zenith angle to distance meter

SD DM = Slope distance from theodolite to distance meter

HD DM = Horizontal distance from theodolite to distance meter

θ = Angle at theodolite between line to distance meter and line to point

AR P = Angle right from backsight to point

ZA P = Zenith angle to point

HD P = Horizontal distance from theodolite station to point

SD P = Slope distance from distance meter to prism

Example:

The theodolite is set up at N100, E150 at elevation 3476.50 feet and instrument height above ground of 4.45 feet. The reference azimuth taken from a backsight is 107°45'22". The distance meter is set up and located from the theodolite by an angle right from the reference of 65°27'18", a slope distance of 10.71 feet and a zenith angle of 89°13'54". The point is located from the theodolite by a angle right from the reference of 75°14'09" with a zenith angle of 90°48'38". The distance meter measures a slope distance to the point of 157.36 feet. The height of the reflector is 5.66 feet. What are the coordinates and elevation of the point?





Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 014	
[XEQ] [ALPHA] 2IRS [ALPHA]	NT ↑ ET
100 [ENTER↑] 150 [F]	ELT=?
3476.5 [G]	HIT=?
4.45 [H]	AZ=?
107.4522 [I]	DIST. METER
	AR ↑ SD ↑
65.2718 [ENTER↑] 10.71 [ENTER↑]	
89.1354 [A]	POINT
	AR ↑ SD ↑
75.1409 [ENTER↑] 157.36 [ENTER↑]	
90.4838 [B]	HIP=?
5.66 [C]	COMPUTE?
[E]	HD=167.8827
[R/S]*	AZ=182.5931
[R/S]*	NP= -67.6539
[R/S]*	EP=141.2373
[R/S]*	ELP=3,472.9148

*Press [R/S] if printer is not attached.

User Instructions

SIZE: 014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Initialize program.		[XEQ] 2IRS	NT ↑ ET
3	Input coordinates of theodolite.	NT	[ENTER↑]	
		ET	[F]	ELT=?
4	Input elevation of theodolite.	ELT	[G]	HIT=?
5	Input height of instrument for theodolite.	HIT	[H]	AZ=?
6	Input reference azimuth and go to step 7	AZ	[I]	DIST. METER AR ↑ SD ↑Δ
	or		[R/S]	BRG ↑ QD
	input reference bearing	BRG	[ENTER↑]	
	and quadrant.	QD	[J]	DIST.METER AR ↑ SD ↑Δ
7.	Input angle right from reference azimuth			
	to dist. meter.	AR(D.MS)	[ENTER↑]	
	Input slope distance from theodolite to			
	distance meter.	SD	[ENTER↑]	
	Input zenith or vertical angle from			
	theodolite to distance meter.	VA/ZA (D.MS)	[A]	POINT AR ↑ SD ↑Δ
8.	Input angle right from reference azimuth			
	to unknown pt.	AR(D.MS)	[ENTER↑]	
	Input slope distance from meter to point	SD	[ENTER↑]	
	Input vertical or zenith angle from			
	theodolite to point.	VA/ZA (D.MS)	[B]	HIP=?
9	Input height of instrument for reflector			
	at point.	HIP	[C]	COMPUTE?
10	Compute horizontal distance and azimuth			
	from theodolite to point, coordinates and			
	elevation of point.		[E]	HD=( )
			[R/S]*	AZ=( )

User Instructions

SIZE: 014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
			[R/S]*	NP=( )
			[R/S]*	EP=( )
			[R/S]*	ELP=( )
11	If the above is to be computed for different			
	inputs, go to the appropriate step to			
	change input then go to step 10.			
	*Press [R/S] if printer is not attached.			

Program Listings

```
01*LBL "2IR
S"
02 SF 27
03 SF 21
04 CF 00
05 FIX 4
06 DEG
07 "NT ↑ ET
"
08 PROMPT
09*LBL F
10 STO 01
11 X<>Y
12 STO 00
13 "ELT=?"
14 PROMPT
15*LBL G
16 STO 02
17 "HIT=?"
18 PROMPT
19*LBL H
20 STO 03
21 CF 22
22 "AZ=?"
23 PROMPT
24 FS?C 22
25 GTO I
26 "BRG ↑ Q
D"
27 PROMPT
28*LBL J
29 X<>Y
30 HR
31 X<>Y
32 ENTER↑
33 ENTER↑
34 2
35 /
36 INT
37 180
38 *
39 X<>Y
40 LASTX
41 *
42 COS
43 R↑
44 *
45 -
46 HMS
47*LBL I
48 HR
```

Initialization

NT, ET input

ELT input

HIT input

Input AZ?

Input BRG, QD
Change to AZ

AZ
REF AZ input

```
49 STO 04
50 "DIST. M
ETER"
51 XEQ 10
52 PROMPT
53*LBL A
54 CF 00
55*LBL 02
56 HR
57 X<>Y
58 P-R
59 X>Y?
60 X<>Y
61 R-P
62 FS? 00
63 RTN
64 STO 07
65 RDN
66 STO 06
67 RDN
68 HR
69 STO 05
70 "POINT"
71 XEQ 10
72 PROMPT
73*LBL B
74 SF 00
75 XEQ 02
76 STO 10
77 RDN
78 STO 09
79 RDN
80 HR
81 STO 08
82 "HIP=?"
83 PROMPT
84*LBL C
85 STO 11
86 "COMPUTE
?"
87 PROMPT
88*LBL E
89 RCL 05
90 RCL 06
91 SIN
92 P-R
93 RCL 08
94 RCL 09
95 SIN
96 P-R
97 X<>Y
```

Distance meter input

ANG

Vertical angle

Distance meter?

Point inputs

HIP inputs

Compute HD, AZ, NP, EP, ELP

Program Listings

```
 98 R↑
 99 *
100 RDN
101 *
102 R↑
103 +
104 RCL 06
105 COS
106 RCL 09
107 COS
108 *
109 +
110 STO 12
111 RCL 07
112 *
113 ENTER↑
114 X↑2
115 RCL 07
116 X↑2
117 -
118 RCL 10
119 X↑2
120 +
121 SQRT
122 +
123 RCL 09
124 SIN
125 *
126 STO 13
127 ADV
128 "HD="
129 ARCL X
130 AVIEW
131 RCL 08
132 RCL 04
133 +
134 1
135 P-R
136 R-P
137 RDN
138 360
139 X<>Y
140 X<0?
141 +
142 HMS
143 "AZ="
144 ARCL X
145 AVIEW
146 LASTX
147 RCL 13
148 P-R
149 RCL 00
150 +
151 ADV
152 "NP="
153 ARCL X
154 AVIEW
155 X<>Y
156 RCL 01
157 +
158 "EP="
159 ARCL X
160 AVIEW
161 RCL 09
162 1
163 P-R
164 X<>Y
165 R-P
166 RDN
167 TAN
168 RCL 13
169 *
170 RCL 02
171 +
172 RCL 03
173 +
174 RCL 11
175 -
176 "ELP="
177 ARCL X
178 PROMPT
179 STOP
180◆LBL 10
181 FC? 55
182 CF 21
183 AVIEW
184 PSE
185 SF 21
186 "AR ↑ SD
 ↑ ∠"
187 RTN
188 END
```

ELP

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	NT	50	
	ET		
	ELT		
	HIT		
	REF AZ		
05	AR DM	55	
	ZA DM		
	SD DM		
	AR P		
	#ZAP		
10	SD P	60	
	HIP		
	COS θ		
	HD		
15		65	

KEYBOARD OVERLAY



STATUS

SIZE 014 TOT. REG. 60 USER MODE

ENG ____ FIX 4 SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
0	C	Point	Distance meter
21	C	Printer Enable	Printer Disable
27	C	USER On	USER Off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input for DM	A	Input NT,ET	F
Input for point	B	Input ELT	G
Input HIP	C	Input HIT	H
Compute values	E	Input AZ	I
		Input BRG, QD	J

TWO INSTRUMENT RADIAL SURVEY

PROGRAM REGISTERS NEEDED: 46

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 - 4)

ROW 2 (4 - 8)

ROW 3 (9 - 15)

ROW 4 (15 - 21)

ROW 5 (21 - 26)

ROW 6 (26 - 30)

ROW 7 (31 - 41)

ROW 8 (42 - 50)

ROW 9 (50 - 53)

ROW 10 (53 - 63)

ROW 11 (64 - 71)

ROW 12 (71 - 78)

ROW 13 (79 - 85)

ROW 14 (86 - 89)

ROW 15 (90 - 102)

ROW 16 (103 - 115)

ROW 17 (116 - 128)

ROW 18 (128 - 137)

ROW 19 (138 - 144)

ROW 20 (145 - 153)

ROW 21 (154 - 162)

ROW 22 (163 - 175)

ROW 23 (176 - 182)

ROW 24 (182 - 186)

ROW 25 (186 - 188)

EDM SLOPE REDUCTION

This program is designed for reducing slope distances measured with an Electronic Distance Meter to a horizontal distance at sea level (HD SL) and a horizontal distance at the instrument station elevation (HD I). Corrections are made for curvature of the earth and for refraction of light (coefficient of refraction 0.071). A radius of 20,906,000 feet is used for the earth.

Equations:

$$ZA' = ZA - c + r + \sin^{-1}\left[\frac{\Delta \sin(ZA - 2c + r)}{SD}\right] \qquad c = \sin^{-1}\left(\frac{(SD)\sin(ZA)}{2R}\right)$$

$$HD\ I = \left(\frac{SD}{\cos c}\right)\sin(ZA' - c) \qquad r = (19 \times 10^{-8})(SD)\sin(ZA)$$

$$HD\ SL = \left(\frac{SD}{\cos c}\right)\sin(ZA' - c)\left[\frac{R}{R + EI}\right] \qquad HD\ SL = (HD\ I)\left[\frac{R}{R + EI}\right]$$

$$\Delta EL = \left(\frac{SD}{\cos c}\right)\cos(ZA') + (HDM - HRFT)$$

$$HD\ I = \left[\frac{(SD)^2 - [(E\ TGT) + (H\ RFT) - EI - (H\ DM)]^2}{[R + EI + (H\ DM)]\,[R + (E\ TGT) + (H\ RFT)]}\right]^{\frac{1}{2}} (R + EI)$$

where:

ZA' = Zenith angle corrected for c, r, and Δ.
ZA = Uncorrected zenith angle.
c = Zenith angle correction due to earth's curvature in degrees.
r = Zenith angle correction due to refraction in degrees.
Δ = (Height of distance meter)-(height of reflector)-(height of theodolite)+ (height of target).
SD = Slope distance, distance meter to reflector.
HD I = Horizontal distance at distance meter station elevation.
HD SL = Horizontal distance at sea level.
R = Radius of earth = 20,906,000 ft.
EI = Elevation of instrument station.
ΔEL = Difference in elevation between distance meter station and target station.
H DM = Height of distance meter above ground.
H RFT = Height of reflector above ground
E TGT = Elevation of target station.

Example:

SD = 10,000.0000
EL I = 5,000.0000
H DM = 5.12
H RFT = 5.75
H TH = 5.96
H TGT = 5.34
ZA = 85°30'20"



Using the values shown in the example, reduce the slope distance to a horizontal distance.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 014	
[XEQ] [ALPHA] EDM [ALPHA]	SD=?
10000 [A]	EL I=?
5000 [B]	H DM=?
5.12 [C]	H RFT=?
5.75 [D]	∠=?
85.302 [H]	H TH=?
5.96 [I]	H TGT=?
5.34 [J]	CHG EL=786.2840
[R/S]*	EL TGT=5,786.2840
[R/S]*	HD SL=9,966.4190
[R/S]*	HD I=9,968.8026

Or, if the elevation of the target (EL TGT=5786.2840) were known, after inputting the height of the reflector (H RFT) above, do the following:

	∠=?
[R/S]	EL TGT=?
5786.284 [E]	CHG EL=786.2840
[R/S]*	EL TGT=5,786.2840
[R/S]*	HD SL=9,966.4166
[R/S]*	HD I=9,968.8002

HD SL and HD I calculated here are slightly different from those calculated one step earlier. This results from inputting EL TGT rounded to 7 places instead of 10 places.

*Press [R/S] if printer is not attached.

User Instructions

SIZE:014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load Program and begin EDM Slope Reduction program.		[XEQ] EDM	SD=?
2.	Input the slope distance, the elevation at	SD	[A]	EL I=?
	the instrument station, the height of the	EL I	[B]	H DM=?
	distance meter and the height of the	H DM	[C]	H RFT=?
	reflector.	H RFT	[D]	∠=?
3.	Input the zenith angle or the vertical			
	angle, if either is known, and go to step 4	(D.MS)	[H]	H TH=?
	Or, if the elevation at the target station			
	is known, press [R/S] without data entry,			
	and go to step 5.		[R/S]	EL TGT=?
4.	Input the height of the theodolite and			H TH=?
	the height of the target, then go to	H TH	[I]	H TGT=?
	step 6.	H TGT	[J]	(see step 6)
5.	Input the elevation of the target and go			EL TGT=?
	to step 6.	EL TGT	[E]	(see step 6)
6.	The results are calculated and displayed			
	as follows:			
	The change in elevation between the			
	instrument and the target.			CHG EL=
	The elevation at the target station.		[R/S]*	EL TGT=
	The horizontal distance at sea level.		[R/S]*	HD SL=
	The horizontal distance at the instru-			
	ment station elevation.		[R/S]*	HD I=
	*Press [R/S] if printer is not attached.			

Program Listings

```
01*LBL "EDM
"
02 DEG
03 FIX 4
04 SF 21
05 SF 27
06 20906 E3
07 STO 10
08 "SD=?"
09 PROMPT
10*LBL A
11 STO 00
12 "EL TGT"
13 ASTO 13
14 "EL I=?"
15 PROMPT
16*LBL B
17 STO 01
18 "H DM=?"
19 PROMPT
20*LBL C
21 STO 02
22 "H RFT=?
"
23 PROMPT
24*LBL D
25 STO 03
26 CF 22
27 "∠=?"
28 PROMPT
29 FC? 22
30 GTO 01
31*LBL H
32 HR
33 1
34 P-R
35 X>Y?
36 X<>Y
37 R-P
38 RDN
39 STO 06
40 SIN
41 RCL 00
42 *
43 STO 07
44 "H TH=?"
45 PROMPT
46*LBL I
47 STO 05
48 "H TGT=?
"
```

Initialize

Input SD

Input EL I

Input H DM

Input H RFT

&

Input ZA/VA

Input

Input H TH

Input H TGT and compute CHG EL, EL TGT, HD SL, HD I

```
49 PROMPT
50*LBL J
51 STO 04
52 RCL 07
53 RCL 10
54 /
55 2
56 /
57 ASIN
58 STO 08
59 RCL 07
60 19 E-8
61 *
62 STO 07
63 RCL 08
64 2
65 *
66 -
67 RCL 06
68 +
69 SIN
70 RCL 02
71 RCL 03
72 -
73 RCL 05
74 -
75 RCL 04
76 +
77 *
78 RCL 00
79 /
80 ASIN
81 RCL 08
82 -
83 RCL 07
84 +
85 RCL 06
86 +
87 STO 09
88 RCL 08
89 -
90 SIN
91 RCL 00
92 RCL 08
93 COS
94 /
95 STO 12
96 *
97 STO 11
98 RCL 10
```

Curvature correction

Program Listings

```
 99 *
100 RCL 10
101 RCL 01
102 +
103 /
104 RCL 09
105 COS
106 RCL 12
107 *
108 RCL 02
109 RCL 03
110 -
111 +
112 RCL 01
113 X<>Y
114 +
115 LASTX
116 GTO 02
117◆LBL 01
118 CLA
119 ARCL 13
120 "⊢=?"
121 PROMPT
122◆LBL E
123 STO 06
124 RCL 00
125 X↑2
126 X<>Y
127 RCL 03
128 +
129 RCL 01
130 -
131 RCL 02
132 -
133 X↑2
134 -
135 RCL 10
136 RCL 01
137 +
138 RCL 02
139 +
140 /
141 RCL 10
142 RCL 06
143 +
144 RCL 03
145 +
146 /
147 SQRT
148 RCL 10
149 RCL 01
```

Prompt for EL TGT

Compute Chg EL EL TGT, HD SL, HD I

```
150 +
151 *
152 STO 11
153 RCL 10
154 *
155 RCL 10
156 RCL 01
157 +
158 /
159 RCL 06
160 RCL 01
161 -
162 RCL 06
163 X<>Y
164◆LBL 02
165 "CHG EL=
"
166 ARCL X
167 AVIEW
168 CLA
169 ARCL 13
170 "⊢="
171 ARCL Y
172 AVIEW
173 "HD SL="
174 ARCL Z
175 AVIEW
176 "HD I="
177 ARCL 11
178 AVIEW
179 STOP
180 END
```

Output routine

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	SD	50	
	EI		
	H DM		
	H RFT		
	H TGT		
05	H TH	55	
	ZA, ETGT		
	r		
	c		
	ZA'		
10	R	60	
	HD I		
	SD/cos c		
	ALPHA DATA		
15		65	
20			
25			
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 014 TOT. REG. 055 USER MODE

ENG ____ FIX 4 SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Stop on aview	
22	C	Numeric input	No numeric input
27	S	USER mode	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input SD	A	Input Δ	H
Input EL I	B	Input H TH	I
Input H DM	C	Input H TGT	J
Input H RFT	D	and compute	
Input EL TGT	E	slope reduction	
and compute			

EDM SLOPE REDUCTION

PROGRAM REGISTERS NEEDED: 42

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 – 5)

ROW 2 (5 – 8)

ROW 3 (8 – 13)

ROW 4 (13 – 18)

ROW 5 (18 – 22)

ROW 6 (22 – 27)

ROW 7 (27 – 35)

ROW 8 (36 – 44)

ROW 9 (44 – 48)

ROW 10 (48 – 59)

ROW 11 (60 – 68)

ROW 12 (69 – 81)

ROW 13 (82 – 94)

ROW 14 (95 – 107)

ROW 15 (108 – 119)

ROW 16 (119 – 127)

ROW 17 (128 – 140)

ROW 18 (141 – 153)

ROW 19 (154 - 165)

ROW 20 (165 - 170)

ROW 21 (170 - 174)

ROW 22 (174 - 180)

ROW 23 (180 - 180)

STADIA REDUCTION

This program is designed to compute the elevation (EL CH) and horizontal distance (HD) for points located by stadia observations. The required information consists of station elevation (EL) height of instrument (HI), rod reading (ROD) if different from HI, vertical or zenith angle (∠), and rod interval (INTV). Stadia constants are contained in the program and may be changed as desired.

Equations:

$$HD = K\ (INTV)\ \cos^2\ (VA) + C\ \cos\ (VA)$$

$$\Delta EL = \frac{K\ (INTV)}{2}\ \sin\ [2(VA)] + C\ \sin\ (VA)$$

Where:

HD = Horizontal distance

K = Stadia interval factor

INTV = Rod interval

VA = Vertical angle of line of sight (0° is horizontal)

C = Stadia constant

ΔEL = Change in elevation



Example:

EL = 491.0 HI = 5.2

ROD	∠	INTV
5.2	75°50'00"	1.31
8.1	93°18'00"	3.32
8.8	90°00'00"	4.06

Reduce the stadia observations given above to elevation and distance for each point.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] STARED [ALPHA]	EL ↑ HI
491 [ENTER↑] 5.2 [A]	ROD=?
[B]	∠=?
75.5 [C]	INTV=?
1.31 [D]	ROD=5.2000
[R/S]*	∠=75.5000
[R/S]*	INTV=1.3100
[R/S]*	HD=123.1532
[R/S]*	EL CH=522.0863
8.1 [B]	∠=?
93.18 [C]	INTV=?
3.32 [D]	ROD=8.1000
[R/S]*	∠=93.1800
[R/S]*	INTV=3.3200
[R/S]*	HD=330.8999
[R/S]*	EL CH=469.0204
8.8 [B]	∠=?
90 [C]	INTV=?
4.06 [D]	ROD=8.8000
[R/S]*	∠=90.0000
[R/S]*	INTV=4.0600
[R/S]*	HD=406.0000
[R/S]*	EL CH=487.4000

*Press [R/S] if printer is not attached.

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize program		[XEQ] STARED	EL ↑ HI
3	Change the stored constants as desired:			
	stadia interval factor (100)	K	[STO] 05	
	stadia constant (0)	C	[STO] 06	
4	Input elevation	EL	[ENTER↑]	
	and height of instrument	HI	[A]	ROD=?
5	Input rod reading if different from HI	ROD	[B]	Δ=?
	otherwise just press [B]		[B]	Δ=?
6	Input vertical or zenith angle	VA/ZA (D.MS)	[C]	INTV=?
7	Input rod interval and compute distance			
	and elevation	INTV	[D]	ROD=( )
			[R/S]*	Δ=( )
			[R/S]*	INTV=( )
			[R/S]*	HD=( )
			[R/S]*	EL CH=( )
8	For a new case, go to step 4. Skip step			
	4, 5 or 6 if any of those inputs have not			
	changed.			
	*Press [R/S] if printer is not attached.			

Program Listings

```
 01*LBL "STA
RED"
 02 DEG
 03 FIX 4
 04 SF 21
 05 SF 27
 06 1 E2
 07 STO 05
 08 0
 09 STO 06
 10 STO 07
 11 "EL ↑ HI
"
 12 PROMPT
 13*LBL A
 14 X<>Y
 15 STO 00
 16 X<>Y
 17 STO 01
 18 STO 02
 19 "ROD=?"
 20 PROMPT
 21*LBL B
 22 STO 02
 23 "∠=?"
 24 PROMPT
 25*LBL C
 26 STO 07
 27 HR
 28 1
 29 P-R
 30 X<=Y?
 31 X<>Y
 32 R-P
 33 RDN
 34 STO 03
 35 "INTV=?"
 36 PROMPT
 37*LBL D
 38 STO 04
 39 RCL 03
 40 RCL 06
 41 P-R
 42 RCL 03
 43 COS
 44 X↑2
 45 RCL 05
 46 *
 47 RCL 04
 48 *
 49 +
```

Initialization

EL, HI input

ROD = HI

ROD input

◁ input

Change to vert. ◁

INTV input

```
 50 X<>Y
 51 RCL 03
 52 2
 53 *
 54 SIN
 55 2
 56 /
 57 RCL 05
 58 *
 59 RCL 04
 60 *
 61 +
 62 RCL 01
 63 +
 64 RCL 02
 65 ADV
 66 "ROD="
 67 ARCL X
 68 AVIEW
 69 -
 70 RCL 00
 71 +
 72 RCL 07
 73 "∠="
 74 ARCL X
 75 AVIEW
 76 RCL 04
 77 "INTV="
 78 ARCL X
 79 AVIEW
 80 ADV
 81 R↑
 82 "HD="
 83 ARCL X
 84 AVIEW
 85 R↑
 86 "EL CH="
 87 ARCL X
 88 PROMPT
 89 END
```

Output routine

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	EL	50	
	HI		
	ROD		
	ANG		
	INTV		
05	K	55	
	C		
	A		
10		60	
15		65	
20			
25			
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 008 TOT. REG. 31 USER MODE
ENG ____ FIX 4 SCI ____ ON ____ OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	C	Printer Enable	Printer Disable
27	C	Set USER Mode	Clear USER Mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input EL, HI	A		
Input ROD	B		
Input ∡	C		
Input INTV & compute	D		

STADIA REDUCTION

PROGRAM REGISTERS NEEDED: 23

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 – 3)

ROW 2 (4 – 11)

ROW 3 (11 – 17)

ROW 4 (18 – 23)

ROW 5 (23 – 33)

ROW 6 (34 – 39)

ROW 7 (40 – 52)

ROW 8 (53 – 65)

ROW 9 (66 – 73)

ROW 10 (73 – 78)

ROW 11 (78 – 86)

ROW 12 (86 – 89)

THREE WIRE LEVELING

This program is designed to compute elevations for a line of wire levels. The required information consists of upper (U), center (C) and lower (L) stadia hair readings for the backsight (BS) and foresight (FS). The elevation is computed by averaging these and the difference between the half stadia intervals is output as a check. The backsight and foresight distances are accumulated and can be output if desired. The program contains a stadia constant which may be changed as desired.

Equations:

$$EL_2 = EL_1 + \frac{(U + C + L)BS}{3} - \frac{(U + C + L)FS}{3}$$

$$\text{Check} = (U - C) - (C - L)$$

$$\Sigma BS = K[\Sigma(U-L)BS]$$

$$\Sigma FS = K[\Sigma(U-L)FS]$$

$$\text{Total distance} = \Sigma BS + \Sigma FS$$

where:

EL_2 = Foresight elevation

EL_1 = Backsight elevation

U, C, L = Upper, center and lower stadia hair readings

BS, FS = Backsight, foresight readings

ΣBS, ΣFS = Sum of backsight, foresight readings

$\Sigma(U-L)$ = Sum of differences between upper and lower stadia hair readings



Example:

EL = 5280.0

Station	BS U, C, L	FS U, C, L
1	8.266	3.491
	8.105	3.320
	7.940	3.152
2	8.119	5.221
	7.329	4.435
	6.535	3.654

Compute the elevations using the stadia data given above.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] 3WL [ALPHA]	EL1=?
5280 [A]	BACKSIGHT RD
	U↑ C↑ L=?
8.266 [ENTER↑] 8.105 [ENTER↑]	
7.94 [B]	BACKSIGHT RD
[R/S]*	U=8.2660
[R/S]*	C=8.1050
[R/S]*	L=7.9400
[R/S]*	CHECK=0.0040
[R/S]*	FORESIGHT RD
	U↑ C↑ L=?
3.491 [ENTER↑] 3.32 [ENTER↑]	
3.152 [C]	FORESIGHT RD
[R/S]*	U=3.4910
[R/S]*	C=3.3200
[R/S]*	L=3.1520
[R/S]*	CHECK=0.0030
[R/S]*	EL2=5,284.7827
[R/S]*	NEXT STA/CMP
8.119 [ENTER↑] 7.329 [ENTER↑]	
6.535 [B]	BACKSIGHT RD
[R/S]*	U=8.1190
[R/S]*	C=7.3290
[R/S]*	L=6.5350
[R/S]*	CHECK=0.0040
[R/S]*	FORESIGHT RD
	U↑ C↑ L=?
5.221 [ENTER↑] 4.435 [ENTER↑]	
3.654 [C]	FORESIGHT RD
[R/S]*	U=5.2210
[R/S]*	C=4.4350
[R/S]*	L=3.6540
[R/S]*	CHECK=0.0050

Keystrokes:	Display:
[R/S]*	EL2=5,287.6737
[R/S]*	NEXT STA/CMP
[D]	BS DIST=191.0000
[R/S]*	FS DIST=190.6000
[R/S]*	TOT DIST=381.6000

*Press [R/S] if printer is not attached.

User Instructions

				SIZE: 008
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize program		[XEQ] 3WL	EL1=?
3	Change stored constant if desired:			
	stadia constant (100.00)	K	[STO] 05	
4	Input beginning elevation	EL_1	[A]	BACKSIGHT RD
				U↑ C↑ L=?
5	Input backsight rod readings:			
	upper reading	U	[ENTER↑]	
	center reading	C	[ENTER↑]	
	lower reading	L	[B]	BACKSIGHT RD
			[R/S]*	U=( )
			[R/S]*	C=( )
			[R/S]*	L=( )
			[R/S]*	CHECK=( )
			[R/S]*	FORESIGHT ROD
				U↑ C↑ L=?
6	Input foresight rod readings:			
	upper reading	U	[ENTER↑]	
	center reading	C	[ENTER↑]	
	lower reading	L	[C]	FORESIGHT ROD
			[R/S]*	U=( )
			[R/S]*	C=( )
			[R/S]*	L=( )
			[R/S]*	CHECK=( )
			[R/S]*	EL2=( )
			[R/S]*	NEXT STA/CMP
7	Go to step 5 for next station input or			

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	compute backsight distance foresight			
	distance and total distance.		[D]	BS DIST=( )
			[R/S]*	FS DIST=( )
			[R/S]*	TOT DIST=( )
*	Press [R/S] if printer is not attached.			

Program Listings

```
 01*LBL "3WL
"
 02 SF 21
 03 SF 27
 04 FIX 4
 05 1 E2
 06 STO 05
 07 0
 08 STO 06
 09 STO 07
 10 "EL1=?"
 11 PROMPT
 12*LBL A
 13 STO 00
 14*LBL 03
 15 "BACK"
 16 XEQ 01
 17 PROMPT
 18*LBL B
 19 0
 20 STO 04
 21 CF 00
 22 XEQ 00
 23 RCL 01
 24 RCL 03
 25 -
 26 ST+ 06
 27 "FORE"
 28 XEQ 01
 29 PROMPT
 30*LBL C
 31 -1
 32 ST* 04
 33 SF 00
 34 XEQ 00
 35 RCL 01
 36 RCL 03
 37 -
 38 ST+ 07
 39 RCL 00
 40 RCL 04
 41 3
 42 /
 43 -
 44 "EL2="
 45 ARCL X
 46 AVIEW
 47 STO 00
 48 "NEXT ST
A/CMP"
 49 PROMPT
```

Initialization

EL_1 input

BS inputs

BS stadia

FS inputs

```
 50 GTO 03
 51*LBL 00
 52 RDN
 53 STO 03
 54 ST+ 04
 55 RDN
 56 STO 02
 57 ST+ 04
 58 RDN
 59 STO 01
 60 ST+ 04
 61 RCL 02
 62 -
 63 RCL 02
 64 RCL 03
 65 -
 66 -
 67 RCL 01
 68 RCL 02
 69 RCL 03
 70 R↑
 71 ABS
 72 ADV
 73 "BACK"
 74 FS? 00
 75 "FORE"
 76 XEQ 02
 77 AVIEW
 78 "U="
 79 ARCL T
 80 AVIEW
 81 "C="
 82 ARCL Z
 83 AVIEW
 84 "L="
 85 ARCL Y
 86 AVIEW
 87 "CHECK="
 88 ARCL X
 89 AVIEW
 90 ADV
 91 RTN
 92*LBL D
 93 RCL 06
 94 RCL 05
 95 *
 96 ENTER↑
 97 ENTER↑
 98 RCL 07
 99 RCL 05
100 *
```

Compute stadia

L

C

U

Output U, C, L, check

BS, FS, total

Program Listings

```
101 +
102 LASTX
103 X<>Y
104 0
105 RDN
106 "BS DIST
="
107 ARCL Z
108 AVIEW
109 "FS DIST
="
110 ARCL Y
111 AVIEW
112 "TOT DIS
T="
113 ARCL X
114 AVIEW
115 STOP
116◆LBL 01
117 XEQ 02
118 CF 21
119 AVIEW
120 PSE
121 SF 21
122 "U↑ C↑ L
=?"
123 RTN
124◆LBL 02
125 "⊢SIGHT
RD"
126 RTN
127 END
```

Data labeling

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	U		
	C		
	L		
	U+C+L		
05	K	55	
	$\Sigma(U-L)$		
	$Z(U-L)^{B}_{F}$		
10		60	
15		65	
20			
25			
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 008 TOT. REG. 47 USER MODE
ENG ____ FIX 4 SCI ____ ON X OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Foresight inputs	Backsight inputs
21	C	Printer Enable	Printer Disable
27	C	User On	User Off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input EL_1	A		
Input BS Readings	B		
Input FS Readings	C		
Compute	D		

THREE WIRE LEVELING

PROGRAM REGISTERS NEEDED: 39

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 – 4)

ROW 2 (5 – 10)

ROW 3 (11 – 16)

ROW 4 (17 – 25)

ROW 5 (26 – 30)

ROW 6 (31 – 38)

ROW 7 (38 – 45)

ROW 8 (46 – 48)

ROW 9 (48 – 57)

ROW 10 (57 – 68)

ROW 11 (69 – 75)

ROW 12 (75 – 80)

ROW 13 (81 – 87)

ROW 14 (87 – 92)

ROW 15 (93 – 105)

ROW 16 (106 – 109)

ROW 17 (109 – 112)

ROW 18 (112 – 116)

ROW 19 (117 - 122)

ROW 20 (122 - 125)

ROW 21 (125 - 127)

AZIMUTH OF THE SUN

This program was designed to compute the azimuth of the sun and reference mark from a solar observation. Required information consists of observer's latitude, time zone, watch correction, temperature, pressure, time of observation, horizontal angle clockwise from reference mark to sun, and vertical or zenith angle. Declination and hour difference in declination are also required and may be obtained from an ephemeris. If declination is on the increase, the hour difference will have a positive value, if on the decrease, the hour difference will be negative.

Equations:

$$\cos(AZ) = \frac{\sin(DEC_c) - \cos(ZA_c)\sin(LAT)}{\sin(ZA_c)\cos(LAT)}$$

$$DEC_c = DEC + \Delta DEC\ (TIME + COR + TZ)$$

$$ZA_c = ZA + \text{refraction} - \text{parallax}$$

$$\text{Refraction} = \frac{45 \times 10^{-4} P}{273 + T} \tan(ZA)$$

$$\text{Parallax} = 25 \times 10^{-4} \sin(ZA)$$

where:

AZ = Azimuth of sun.
DEC_c = Corrected declination.
ZA_c = Corrected zenith angle
LAT = Latitude
DEC = Declination at 0^{hr} GCT.
ΔDEC = Change in declination per hour.
TIME = Local zone time of observation.
COR = Watch correction.
TZ = Time zone.
ZA = Observed zenith angle.
P = Pressure in millibars.
T = Temperature in °C.

NOTE:

The program accepts temperature in degrees Fahrenheit and pressure in inches of mercury.

Example:

date - August 14, 1979
latitude - 37°49'.5
time zone - 5 hrs.
watch correction - -2'49"
temperature - 66°F
pressure - 28.58" Hg
declination - 14°23'.6 Oh UTC
hour difference - -0'.76



DIRECTION	HORIZONTAL ANGLE	TIME	ZENITH ANGLE
direct	77°53'33"	16h33m30s	62°12'37"
reverse	258°17'01"	16h35m46s	297°21'16"
direct	78°33'44"	16h37m31s	62°59'21"
reverse	258°50'02"	16h39m05s	296°42'14"

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 016	
[XEQ] [ALPHA] AZIM [ALPHA]	LAT=?
3749.5 [F]	TZ ↑ COR=?
5 [ENTER↑] .0249 [CHS] [G]	T ↑ P=?
66 [ENTER↑] 28.58 [H]	DEC ↑ DIF=?
1423.6 [ENTER↑] .76 [CHS] [A]	HA D=?
77.5333 [B]	TIME=?
16.3330 [D]	VA / ZA CHS?
62.1237 [CHS] [E]	S AZ=266.3447
[R/S]*	R AZ=188.4114
258.1701 [C]	TIME=?
16.3546 [D]	VA / ZA CHS?
297.2116 [CHS] [E]	S AZ=266.5551
[R/S]*	R AZ=188.3850
78.3344 [B]	TIME=?
16.3731 [D]	VA / ZA CHS?
62.5921 [CHS] [E]	S AZ=267.1224
[R/S]*	R AZ=188.3840
258.5002 [C]	TIME=?
16.3905 [D]	VA / ZA CHS?
296.4214 [CHS] [E]	S AZ=267.2709
[R/S]*	R AZ=188.3707
[J]	MEAN
[R/S]*	S AZ=267.0233
[R/S]*	R AZ=188.3858

*Press [R/S] if printer is not attached.

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize program		[XEQ] AZIM	LAT=?
3	Input latitude	LAT(DM.m)[1]	[F]	TZ ↑ COR=?
4	Input time zone and	TZ	[ENTER↑]	
	watch correction	COR(H.MS)	[G]	T ↑ P=?
5	Input temperature	T(°F)	[ENTER↑]	
	and pressure	P(in.Hg.)	[H]	DEC ↑ DIF=?
6	Input declination at 0h. UTC	DEC(DM.m)[1]	[ENTER↑]	
	and hour difference	DIF(M.m)[2]	[A]	HA D=?
	or, if declination is decreasing, input			
	negative hour difference	DIF(M.m)[2]	[CHS] [A]	HA D=?
7	Input horizontal angle clockwise from			
	reference to sun for direct reading	HA D(D.MS)	[B]	TIME=?
	or for reverse reading	HA R(D.MS)	[C]	TIME=?
8	Input local zone time of observation	TIME(H.MS)	[D]	VA / ZA CHS?
9	Input vertical angle	VA(D.MS)	[E]	
	or zenith angle. Output will be sun's	ZA(D.MS)	[CHS] [E]	S AZ=( )
	azimuth and Reference azimuth		[R/S]*	R AZ=( )
10	For next sight in same group, go to step 7			
11	Compute mean of sights		[J]	MEAN
			[R/S]*	S AZ=( )
			[R/S]*	R AZ=( )
12	For a new group of sights, press I to			
	clear. If any of the inputs on steps 2-5			
	has changed, input the new value at the			
	appropriate step and then go to step 7.			

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	[1]DM.m = key in degrees, then minutes			
	(2 digits), then decimal point, then			
	tenths of minutes.			
	[2]M.m = key in minutes, then decimal point,			
	then tenths of minutes.			
	*Press [R/S] if printer is not attached.			

Program Listings

```
01*LBL "AZIM"
02 SF 21
03 SF 27
04 FIX 4
05 DEG
06 ΣREG 09
07*LBL I
08 14.008
09 STO 15
10 0
11*LBL 05
12 STO IND 15
13 DSE 15
14 GTO 05
15 "LAT=?"
16 PROMPT
17*LBL F
18 HMS
19 1 E2
20 /
21 HR
22 STO 00
23 0
24 "TZ ↑ COR=?"
25 PROMPT
26*LBL G
27 HR
28 +
29 STO 01
30 "T ↑ P=?"
31 PROMPT
32*LBL H
33 2.983
34 /
35 51
36 *
37 X<>Y
38 460
39 +
40 /
41 STO 02
42 "DEC ↑ DIF=?"
43 PROMPT
44*LBL A
45 60
46 /
47 STO 05
48 X<>Y
49 HMS
50 E2
51 /
52 HR
53 STO 04
54 CF 22
55 "HA D=?"
56 PROMPT
57 FS?C 22
58 GTO B
59 "HA R=?"
60 PROMPT
61*LBL C
62 SF 02
63 -180
64 HMS+
65 GTO 03
66*LBL B
67 CF 02
68*LBL 03
69 HR
70 1
71 P-R
72 R-P
73 RDN
74 360
75 X<>Y
76 X<0?
77 +
78 STO 03
79 "TIME=?"
80 PROMPT
81*LBL D
82 HR
83 STO 06
84 "VA / ZACHS?"
85 PROMPT
86*LBL E
87 HR
88 X>0?
89 GTO 00
90 ABS
91 1
92 P-R
93 X<>Y
94 R-P
95 RDN
```

Initialization

Clear Σ regs

change LAT to DMS

TZ, COR input

T & P input

Input HA D

HA R input
Mark HA R

HA D input
Mark HA D
normalize

Time input

VA (-AZ) input & compute S AZ, R AZ

Program Listings

```
 96◆LBL 00
 97 180
 98 X<>Y
 99 FS?C 02
100 -
101 STO 07
102 RCL 07
103 RCL 07
104 3
105 +
106 12
107 *
108 ATAN
109 -
110 TAN
111 .97
112 *
113 60
114 /
115 RCL 02
116 *
117 +
118 R↑
119 COS
120 25 E-4
121 *
122 +
123 1
124 P-R
125 RCL 01
126 RCL 06
127 +
128 RCL 05
129 *
130 RCL 04
131 +
132 SIN
133 RCL 00
134 SIN
135 R↑
136 *
137 -
138 X<>Y
139 RCL 00
140 COS
141 *
142 /
143 ACOS
144 RCL 06
145 12
146 X<=Y?
147 GTO 02
148 RDN
149 RDN
150◆LBL 01
151 STO 08
152 RCL 03
153 -
154 RCL 08
155 Σ+
156 LASTX
157◆LBL 04
158 HMS
159 ADV
160 "S AZ="
161 ARCL X
162 AVIEW
163 X<>Y
164 HMS
165 "R AZ="
166 ARCL X
167 AVIEW
168 STOP
169◆LBL 02
170 RDN
171 RDN
172 CHS
173 360
174 +
175 GTO 01
176◆LBL J
177 MEAN
178 ADV
179 "MEAN"
180 AVIEW
181 GTO 04
182 .END.
```

VA		
Reverse?		
		Output routine
		Sun's AZ
		P.M. adjust
		Compute mean
	90	
	00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	LAT	50	
	TZ + COR		
	T,P, COR		
	HA		
	DEC		
05	DIF	55	
	TIME		
	VA		
	SUN AZ		
	E		
10	USED	60	
	USED		
	USED		
	USED		
	USED		
15	INDIRECT	65	
20			
25			
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 016 TOT. REG. 63 USER MODE

ENG ____ FIX 4 SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
02	C	HA R input	HA D input
21	C	Printer Enable	Printer Disable
22	C	Numeric Input	
27	C	User On	User Off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input DEC,DIF	A	Input LAT	F
Input HA D	B	Input TZ,COR	G
Input HA R	C	Input T & P	H
Input TIME	D	Clear	I
Input VA/ZA	E	Compute mean	J

AZIMUTH OF THE SUN

PROGRAM REGISTERS NEEDED: 48

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (10 - 15)

ROW 4 (15 - 24)

ROW 5 (24 - 26)

ROW 6 (27 - 32)

ROW 7 (32 - 38)

ROW 8 (38 - 42)

ROW 9 (42 - 50)

ROW 10 (50 - 55)

ROW 11 (56 - 59)

ROW 12 (60 - 66)

ROW 13 (66 - 75)

ROW 14 (76 - 81)

ROW 15 (82 - 84)

ROW 16 (84 - 93)

ROW 17 (94 - 103)

ROW 18 (104 - 113)

ROW 19 (113 - 121)

ROW 20 (122 - 134)

ROW 21 (135 - 146)

ROW 22 (147 - 158)

ROW 23 (159 - 165)

ROW 24 (165 - 172)

ROW 25 (173 - 179)

ROW 26 (179 - 182)

TAPING REDUCTION

This program corrects taped distances for temperature, pull (tension), sag and index. The program contains a series of constants, but these may be changed as desired. Slope distances can also be reduced to horizontal distances with correction. The slope can be indicated either by an elevation difference or by a vertical or zenith angle.

Equations:

$$\text{Correction for tension} = \frac{L}{AE}(P-P_o) \qquad \text{Index correction} = \frac{L\ (CI)}{100}$$

$$\text{Correction for temperature} = L\ (CT)\ (T-T_o) \qquad HD_c = (SD_c)\ \sin\ (ZA) = (SD_c)\ \cos\ (VA)$$

$$\text{Correction for sag} = \frac{W^2L^3}{24\ P^2} \qquad HD_c = \sqrt{(SD_c)^2 - (\Delta EL)^2}$$

where:

L = Distance between supports in feet.

A = Cross section of tape in square inches.

E = Modulus of elasticity for the tape material.

P = Tension on tape in pounds (pull).

P_o = Standard tension in pounds.

CT = Coefficient of thermal expansion for the tape.

T = Temperature at time of reading in °F.

T_o = Standard temperature in °F.

W = Weight of tape per foot in pounds.

CI = Tape correction.

HD_c = Horizontal distance (corrected).

SD_c = Slope distance (corrected).

ZA = Zenith angle.

VA = Vertical angle.

ΔEL = Change in elevation.

Example:

Using the stored constants and a temperature of 88°F and tension of 20 lbs., reduce the following taped measurements:

1) HD = 100.0000

2) SD = 100.0000 ΔEL = -3.7500

Then change the stored standard tension to 35 lbs. and reduce

3) SD = 77.4580 ZA = 88°05'43"

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
XEQ [ALPHA] TPRED [ALPHA]	T ↑ PULL
88 [ENTER ↑] 20 [A]	HD/SD,EL/SD,∠?
100 [B]	HD=100.0000
[R/S]*	HDC=99.9895
[R/S]*	ΣHDC=99.9895
100 [ENTER ↑] 3.75 [CHS] [C]	EL CH=-3.7500
[R/S]*	SD=100.0000
[R/S]*	HDC=99.9191
[R/S]*	ΣHDC=199.9086
35 [STO] 00	35.0000
77.458 [ENTER ↑] 88.0543 [D]	∠=88.0543
[R/S]*	SD=77.4580
[R/S]*	HDC=77.4143
[R/S]*	ΣHDC=277.3229

*Press [R/S] if printer is not attached.

User Instructions

				SIZE: 013
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize program		[XEQ] TPRED	T ↑ PULL
3	Change any stored constants (shown in			
	parenthesis) as desired:			
	standard tension (20 lbs.)	P_o	[STO] 00	
	cross section area (.009 in^2)	A	[STO] 01	
	modulus of elasticity (3.0×10^7 psi)	E	[STO] 02	
	standard temperature (68°F)	T_o	[STO] 03	
	coefficient of thermal expansion			
	(6.45×10^{-6}/°F)	CT	[STO] 04	
	weight (.015 lbs./ft.)	W	[STO] 05	
	index correction (0.0)	CI	[STO] 06	
4	Input temperature	T	[ENTER ↑]	
	and pull	PULL	[A]	HD/SD,EL/SD, ?
5	Input horizontal distance	HD	[B]	HD=( )
			[R/S]*	HDC=( )
			[R/S]*	ΣHDC=( )
5a	or slope distance	SD	[ENTER ↑]	
	and elevation difference	ΔEL	[C]	EL CH=( )
			[R/S]*	SD=( )
			[R/S]*	HDC=( )
			[R/S]*	ΣHDC=( )
5b	or slope distance	SD	[ENTER ↑]	
	and angle	ANG (DMS)	[D]	∠=( )
			[R/S]*	SD=( )
			[R/S]*	HDC=( )
			[R/S]*	ΣHDC=( )

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
6	Repeat steps 4 and 5 for next reduction.			
	If T and Pull do not change, skip step 4.			
*	Press [R/S] if printer is not attached.			

Program Listings

```
 01♦LBL "TPR
ED"
 02 SF 21
 03 SF 27
 04 CF 01
 05 CLRG
 06 FIX 4
 07 20
 08 STO 00
 09 .009
 10 STO 01
 11 3 E7
 12 STO 02
 13 68
 14 STO 03
 15 645 E-8
 16 STO 04
 17 .015
 18 STO 05
 19 "T ↑ PUL
L"
 20 PROMPT
 21♦LBL A
 22 STO 07
 23 RCL 00
 24 -
 25 RCL 01
 26 /
 27 RCL 02
 28 /
 29 STO 08
 30 X<>Y
 31 RCL 03
 32 -
 33 RCL 04
 34 *
 35 ST+ 08
 36 RCL 06
 37  E2
 38 /
 39 ST+ 08
 40 RCL 05
 41 X↑2
 42 24
 43 /
 44 RCL 07
 45 X↑2
 46 /
 47 STO 10
 48 "HD/SD,E
L/SD,∠?"
```

Initialization

T, Pull input

Tension corr.

Temperature corr.

Index corr.

Sag corr.

```
 49 PROMPT
 50♦LBL B
 51 SF 01
 52 CF 02
 53 STO 11
 54 0
 55 STO 12
 56♦LBL 00
 57 RCL 11
 58 ENTER↑
 59 ENTER↑
 60 ENTER↑
 61 *
 62 *
 63 RCL 10
 64 *
 65 X<>Y
 66 RCL 08
 67 *
 68 -
 69 -
 70 FS?C 02
 71 RTN
 72♦LBL 01
 73 ST+ 09
 74 RCL 12
 75 FS? 00
 76 HMS
 77 RCL 11
 78 RCL 09
 79 R↑
 80 X<>Y
 81 "HD="
 82 FS?C 01
 83 GTO 02
 84 "EL CH="
 85 FS?C 00
 86 "∠="
 87 ARCL T
 88 AVIEW
 89 "SD="
 90♦LBL 02
 91 ARCL Z
 92 AVIEW
 93 "HDC="
 94 ARCL Y
 95 AVIEW
 96 "ΣHDC="
 97 ARCL X
 98 PROMPT
 99♦LBL C
```

HD input

Correct SD

Output routine

Output HDC, ΣHDC

SD, ΔEL input

Program Listings

```
100 CF 00
101 STO 12
102 X<>Y
103 STO 11
104 SF 02
105 XEQ 00
106 X↑2
107 RCL 12
108 X↑2
109 -
110 SQRT
111 GTO 01
112◆LBL D
113 SF 00
114 HR
115 STO 12
116 X<>Y
117 STO 11
118 SF 02
119 XEQ 00
120 RCL 12
121 X<>Y
122 P-R
123 X<=Y?
124 X<>Y
125 GTO 01
126 END
```

Δ EL

RTN

HD

SD, ANG input

ANG

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Po	50	
	A		
	E		
	To		
	CT		
05	W	55	
	C1		
	P		
	C or		
	ΣHDC		
10	C or	60	
	SD		
	ZA		
15		65	

KEYBOARD OVERLAY



STATUS

SIZE 013 TOT. REG. 46 USER MODE

ENG ____ FIX 4 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Angle input	not input
01		Entered HD	HD not entered
02	C	LBL 00 subroutine	LBL 00 not sub
21	C	Printer Enable	Printer Disable
27	C	User On	User Off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Enter T, Pull	A		
Enter HD&compute	B		
Enter SD, EL & compute	C		
Enter SD, & compute	D		

TAPING REDUCTION

PROGRAM REGISTERS NEEDED: 33

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 - 3)

ROW 2 (4 - 10)

ROW 3 (11 - 15)

ROW 4 (16 - 19)

ROW 5 (19 - 29)

ROW 6 (30 - 39)

ROW 7 (40 - 48)

ROW 8 (48 - 50)

ROW 9 (50 - 60)

ROW 10 (61 - 72)

ROW 11 (73 - 81)

ROW 12 (81 - 85)

ROW 13 (85 - 91)

ROW 14 (91 - 96)

ROW 15 (96 - 103)

ROW 16 (104 - 112)

ROW 17 (112 - 120)

ROW 18 (121 - 126)

TRIANGLE SOLUTIONS

These programs can be used to find the area, the dimensions of the sides (S_1, S_2, S_3) and the angles (A_1, A_2, A_3) of a triangle.



Simply key in three known values and execute the appropriate program. The calculator will output the values of the sides, the angles, and the area. The order of output is determined by the order of input. If input values are selected in a clockwise order around the triangle, the outputs will also follow a clockwise order around the triangle. The order is as follows:

First side input	(S_1)
Adjacent angle	(A_1)
Adjacent side	(S_2)
Adjacent angle	(A_2)
Adjacent side	(S_3)
Adjacent angle	(A_3)
Area	

Equations:

S_1, S_2, S_3 (all sides of triangle are known)

where $P = (S_1 + S_2 + S_3)/2$

$$A_3 = 2 \cos^{-1} \sqrt{\frac{P(P - S_2)}{S_1 S_3}}$$

$$A_2 = 2 \cos^{-1} \sqrt{\frac{P(P - S_1)}{S_2 S_3}}$$

$$A_1 = \cos^{-1} \left(-\cos(A_3 + A_2)\right)$$

A_3, S_1, A_1 (Two angles and the included side are known)

$$A_2 = \cos^{-1} \left(-\cos(A_3 + A_1)\right)$$

$$S_2 = S_1 \frac{\sin A_3}{\sin A_2}$$

$$S_3 = S_1 \cos A_3 + S_2 \cos A_2$$

S_1, A_1, A_2 (side and following two angles known)

$$A_3 = \cos^{-1} \left(-\cos(A_1 + A_2)\right)$$

Problem has been reduced to the A_3, S_1, A_1 configuration.

S_1, A_1, S_2 (Two sides and included angle are known)

$$S_3 = \sqrt{S_1^2 + S_2^2 - 2\,S_1 S_2 \cos A_1}$$

The problem has been reduced to the S_1, S_2, S_3 configuration.

S_1, S_2, A_2 (Two sides and the adjacent angle known)

$$A_3 = \sin^{-1}\left[\frac{S_2}{S_1}\sin A_2\right]^*$$

$$A_1 = \cos^{-1}\left[-\cos(A_2 + A_3)\right]$$

The problem has been reduced to the A_3, S_1, A_1 configuration.



$$\text{Area} = 1/2\ S_1\ S_3 \sin A_3$$

*Note that two possible solutions exist if S_2 is greater than S_1 and A_3 does not equal 90°. Both possible answer sets are calculated.

Remarks:

- Note that the triangle described by the program does not conform to the standard triangle notation, i.e., A_1 is not opposite S_1.
- Accuracy of solution may degenerate for triangles containing extremely small angles.

Example:

A surveyor is to find the area and dimensions of a triangular land parcel. From point A, the distances to B and C are measured with an electronic distance meter. The angle between AB and AC is also measured. Find the area and other dimensions of the triangle.



This is a side-angle-side problem where: $S_1 = 171.63$, $A_1 = 98°12'$ and $S_2 = 297.35$.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] SAS [ALPHA]	S1=?
171.63 [R/S]	A1=?
98.12 [R/S]	S2=?
297.35 [R/S]	S1=171.6300
[R/S]*	A1=98.1200
[R/S]*	S2=297.3500
[R/S]*	A2=27.4937
[R/S]*	S3=363.9118
[R/S]*	A3=53.5823
[R/S]*	AREA=25,256.2094

*Press [R/S] if printer is not attached.

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and find applicable case in the list below.			
1a	Input indicated values:			
	All sides known		[XEQ] SSS	S1=?
		S_1	[R/S]	S2=?
		S_2	[R/S]	S3=?
		S_3	[R/S]	S1=
1b	Two angles and included side known		[XEQ] ASA	A3=?
		A_3 (H.MS)	[R/S]	S1=?
		S_1	[R/S]	A1=?
		A_1 (H.MS)	[R/S]	S1=?
1c	Two angles and adjacent side known		[XEQ] SAA	S1=?
		S_1	[R/S]	A1=?
		A_1 (H.MS)	[R/S]	A2=?
		A_2 (H.MS)	[R/S]	S1=
1d	Two sides and included angle known		[XEQ] SAS	S1=?
		S_1	[R/S]	A1=?
		A_1 (H.MS)	[R/S]	S2=?
		S_2	[R/S]	S1=
1e	Two sides and adjacent angle known		[XEQ] SSA	S1=?
		S_1	[R/S]	S2=?
		S_2	[R/S]	A2=?
		A_2 (H.MS)	[R/S]	S1=
2	After step 1, the values of the sides and			
	angles are output. The last output is the			
	triangle's area. For the last case (SSA),			
	two possible solutions may exist and both			
	will be output.			

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
			[R/S]*	A1=
			[R/S]*	S2=
			[R/S]*	A2=
			[R/S]*	S3=
			[R/S]*	A3=
			[R/S]*	AREA=
*	Press [R/S] if printer is not attached.			

Program Listings

```
01◆LBL "SSS
"
02 XEQ 06
03 "S3=?"
04 PROMPT
05 STO 04
06◆LBL 05
07 +
08 +
09 2
10 /
11 STO 07
12 X↑2
13 LASTX
14 RCL 02
15 *
16 -
17 RCL 00
18 RCL 04
19 *
20 /
21 SQRT
22 ACOS
23 2
24 *
25 STO 05
26 SIN
27 RCL 00
28 *
29 STO 06
30 RCL 07
31 X↑2
32 LASTX
33 RCL 00
34 *
35 -
36 RCL 02
37 /
38 RCL 04
39 /
40 SQRT
41 ACOS
42 2
43 *
44 STO 03
45 RCL 05
46 XEQ 00
47 STO 01
48 GTO 01
49◆LBL 00
50 +
```

Calculate P

Calculate A_3 knowing S_1, S_2, S_3

$h=S_1 \sin A_3$

Calculate A_2 knowing S_1, S_2, S_3

```
51 COS
52 CHS
53 ACOS
54 RTN
55◆LBL "SAA
"
56 XEQ 07
57 "A2=?"
58 PROMPT
59 HR
60 STO 03
61 RCL 01
62 XEQ 00
63 STO 05
64 GTO 04
65◆LBL 06
66 SF 21
67 DEG
68 FIX 4
69 "S1=?"
70 PROMPT
71 STO 00
72 "S2=?"
73 PROMPT
74 STO 02
75 RTN
76◆LBL 07
77 SF 21
78 DEG
79 FIX 4
80 "S1=?"
81 PROMPT
82 STO 00
83 "A1=?"
84 PROMPT
85 HR
86 STO 01
87 RTN
88◆LBL "ASA
"
89 SF 21
90 "A3=?"
91 PROMPT
92 HR
93 STO 05
94 XEQ 07
95 RCL 05
96◆LBL 04
97 RCL 01
98 XEQ 00
99 STO 03
```

Calculate A_3 knowing S_1, A_1, A_2

Input routines

$y = S_1 \sin A_3$

Program Listings

```
100 RCL 05
101 RCL 00
102 P-R
103 X<>Y
104 STO 06
105 RCL 03
106 1
107 P-R
108 RDN
109 /
110 STO 02
111 R↑
112 *
113 +
114 STO 04
115 GTO 01
116◆LBL "SAS
"
117 XEQ 07
118 "S2=?"
119 PROMPT
120 STO 02
121 P-R
122 RCL 00
123 -
124 R-P
125 STO 04
126 RCL 00
127 RCL 02
128 RCL 04
129 GTO 05
130◆LBL "SSA
"
131 XEQ 06
132 "A2=?"
133 PROMPT
134 HR
135 STO 03
136 SIN
137 RCL 02
138 *
139 RCL 00
140 /
141 ASIN
142 XEQ 03
143 XEQ 04
144 RCL 00
145 RCL 02
146 X<=Y?
147 GTO 02
148 RCL 05
```

x = S_1 cos A_3

h = x

y = sin A_2
x = cos A_2

Calculate S_2 & S_3 knowing A_3, S_1 and A_1

Calculate S_3 knowing S_1, A_1 and S_2

Calculate A_3 knowing S_1, S_2 and A_2

Stop if this is only solution

```
149 COS
150 CHS
151 ACOS
152 XEQ 03
153 GTO 04
154◆LBL 03
155 STO 05
156 RCL 03
157 XEQ 00
158 STO 01
159 RCL 05
160 RTN
161◆LBL 02
162 RDN
163 RDN
164 RTN
165◆LBL 01
166 "S1="
167 ARCL 00
168 AVIEW
169 "A1="
170 RCL 01
171 HMS
172 ARCL X
173 AVIEW
174 "S2="
175 ARCL 02
176 AVIEW
177 "A2="
178 RCL 03
179 HMS
180 ARCL X
181 AVIEW
182 "S3="
183 ARCL 04
184 AVIEW
185 "A3="
186 RCL 05
187 HMS
188 ARCL X
189 AVIEW
190 RCL 06
191 RCL 04
192 *
193 2
194 /
195 "AREA="
196 ARCL X
197 AVIEW
198 END
```

Find secondary angle for alternate solution

Output routine

Area = (S_1, S_3 sin A_3)/2

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	S_1	50	
	A_1		
	S_2		
	A_2		
	S_3		
05	A_3	55	
	h		
	P		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 56 USER MODE

ENG ____ FIX 4 SCI ____ ON ____ OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	C	Printer Enable	Printer Disable

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TRIANGLE SOLUTIONS

PROGRAM REGISTERS NEEDED: 48

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 - 3)

ROW 2 (3 - 14)

ROW 3 (15 - 27)

ROW 4 (28 - 40)

ROW 5 (41 - 50)

ROW 6 (51 - 56)

ROW 7 (56 - 62)

ROW 8 (63 - 69)

ROW 9 (69 - 77)

ROW 10 (77 - 83)

ROW 11 (83 - 88)

ROW 12 (88 - 94)

ROW 13 (94 - 104)

ROW 14 (105 - 116)

ROW 15 (116 - 118)

ROW 16 (118 - 129)

ROW 17 (130 - 132)

ROW 18 (132 - 142)

ROW 19 (142 - 151)

ROW 20 (152 - 159)

ROW 21 (160 - 168)

ROW 22 (169 - 174)

ROW 23 (175 - 182)

ROW 24 (182 - 188)

ROW 25 (188 - 195)

ROW 26 (196 - 198)

TRAVERSE FOR AUTO ADJUST ROUTINES, AUTO ADJUST FOR COMPASS RULE, AUTO ADJUST FOR CRANDALL'S RULE

(These programs require a card reader and one memory module)

The program "Traverse for Auto Adjust Routines" performs traverse computations and permits permanent storage of all traverse data on magnetic cards. The data may then be used to automatically perform traverse adjustments using the two programs: "Auto Adjust for Compass Rule" and "Auto Adjust for Crandall's Rule". After the traverse is completed, the program outputs the latitude error and the departure error.

Reference:

These programs are based on HP User's Library programs by Earl L. Kubaskie, Jr.

Equations:

Traverse Computation for Auto Adjust Routines:

$$\text{Azimuth} = 180 \left\{ \text{INT} \frac{QD}{2} - \text{BRG} \cos [(180)(QD)] \right\}$$

HD = SD sin (zentih angle)

HD = SD cos (vertical angle)

$\text{Latitude}_k = \text{LAT}_k = N_{k+1} - N_k$

$\text{Departure}_k = \text{DEP}_k = E_{k+1} - E_k$

For instance: $\text{LAT}_1 = N_2 - N_1$

For instance: $\text{DEP}_4 = E_5 - E_4$

where:

INT = Integer portion of number (portion to left of decimal point)
QD = Quadrant
BRG = Bearing
HD = Horizontal distance
SD = Slope distance

Compass Rule for latitude and departure course correction:

$$\text{Corrected latitude}_1 = L_1 + \ell_1 = L_1 + \frac{(HD)_1 (ER\ L)}{\Sigma (HD)}$$

$$\text{Corrected departure}_1 = D_1 + d_1 = D_1 + \frac{(HD)_1 (ER\ D)}{\Sigma (HD)}$$

Crandall's Rule for latitude and departure course correction:

$$\text{Corrected Latitude} = L + [(L)(A) + (D)(B)] \left[\frac{L}{(HD)}\right]$$

$$\text{Corrected Departure} = D + [(L)(A) + (D)(B)] \left[\frac{D}{(HD)}\right]$$

$$A = \frac{(ER\ D)\left[\sum\frac{(L)(D)}{(HD)}\right] - (ER\ L)\left[\sum\frac{D^2}{(HD)}\right]}{\left[\sum\frac{D^2}{(HD)}\right]\left[\sum\frac{L^2}{(HD)}\right] - \left[\sum\frac{(L)(D)}{(HD)}\right]^2}$$

$$B = \frac{(ER\ L)\left[\sum\frac{(L)(D)}{(HD)}\right] - (ER\ D)\left[\sum\frac{L^2}{(HD)}\right]}{\left[\sum\frac{D^2}{(HD)}\right]\left[\sum\frac{L^2}{(HD)}\right] - \left[\sum\frac{(L)(D)}{(HD)}\right]^2}$$

Where:

A and B are intermediate values used in the calculations.
(ER D) = Total error in departure.
(ER L) = Total error in latitude.
L = Uncorrected latitude of any course.
D = Uncorrected departure of any course.
(HD) = Uncorrected horizontal distance of any course.
ℓ_1 = Correction to be applied to the uncorrected latitude of the first course.
d_1 = Correction to be applied to the uncorrected departure of the first course.

Angle conventions for azimuths and quadrant bearings are shown below:



Example:



Perform traverse computations with the data given in the sketch and prepare the data for adjustment.

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] TRVAA [ALPHA]	BEG N↑ BEG E
100 [ENTER↑] 200 [A]	N1=100.0000
[R/S]*	E1=200.0000
[R/S]*	COURSE 1
22.1347 [ENTER↑] 4 [B]	AZ=337.4613
[R/S]*	HD OR SD ↑ ∠
152.06 [D]	HD=152.0600
[R/S]*	N2=240.7581
[R/S]*	E2=142.4725
[R/S]*	NEXT COURSE?
.	.
.	.
.	.

After entering the data for course 5 continue with course 6:

265.5750 [C]	AZ=265.5750
[R/S]*	HD OR SD ↑ ◁
286.2 [D]	HD=286.2000
[R/S]*	N6=100.0320
[R/S]*	E6=200.1251
[R/S]*	NEXT COURSE?
100 [ENTER↑] 200 [E]	LAT ERR=0.0320
[R/S]*	DEP ERR=0.1251
	RDY 01 OF 02
Read in side 1 of card	RDY 02 OF 02
Read in side 2 of card	0.0000

Now the data is set up for adjustment by the two programs: Auto Adjust for Compass Rule and Auto Adjust for Crandall's Rule. The results output by both of these programs are corrected azimuths, horizontal distances, and coordinates.

Use the data from the program Traverse Computation and storage in the program Auto Adjust for Compass Rule (Auto Adjust for Crandall's Rule works similarly).

*[R/S] is required if printer is not used.

Key in program:

[XEQ] [ALPHA] AACMP [ALPHA]	LAST CARD
	CARD
Read in side 1 & 2 of card	N1=100.0000
[R/S]*	E1=200.0000
[R/S]*	AZ1=337.4547
[R/S]*	HD1=152.0626
[R/S]*	N2=240.7533
[R/S]*	E2=142.4541
[R/S]*	AZ2=94.2952
[R/S]*	HD2=238.9218
[R/S]*	N3=222.0166
[R/S]*	E3=380.6401
[R/S]*	AZ3=78.4248
[R/S]*	HD3=193.4910
[R/S]*	N4=259.8859
[R/S]*	E4=570.3891
[R/S]*	AZ4=211.1618
[R/S]*	HD4=163.4846
[R/S]*	N5=120.1531
[R/S]*	E5=485.5248
[R/S]*	AZ5=265.5745
[R/S]*	HD5=286.2352
[R/S]*	N6=100.0000
[R/S]*	E6=200.0000

*[R/S] is required if printer is not used.

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Traverse program			
2	Initialize program		[XEQ] TRVAA	BEG N↑ BEG E
3	Input beginning coordinates	BEG N	[ENTER↑]	
		BEG E	[A]	N1=
				E1=
				COURSE 1
4	For bearing/azimuth traverse go to step 5.			
	For field angle traverse go to step 8.			
5	Input bearing	BRG (D.MS)	[ENTER↑]	
	and quadrant code	QD	[B]	AZ=(D.MS)
			[R/S]*	HD OR SD ↑∠
	or			
5a	input azimuth	AZ (D.MS)	[C]	AZ=(D.MS)
			[R/S]*	HD OR SD ↑∠
6	Input horizontal distance	HD	[D]	HD=
			[R/S]*	N#=
			[R/S]*	E#=
			[R/S]*	NEXT COURSE?
	or			
6a	input slope distance and	SD	[ENTER↑]	
	vertical or zenith angle	∠ (D.MS)	[I]	HD=
			[R/S]*	N#=
			[R/S]*	E#=
			[R/S]*	NEXT COURSE?
				or
	On 10th course, display shows "RDY 01 OF 02"			RDY 01 OF 02
	In this case go to step 10c			

*[R/S] required if printer is not used.

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
7	Go to step 4 for further courses			
	or			
	go to step 12 if traverse is completed			
8	Input reference azimuth:			
8a	away from beginning point	R.AZ- (D.MS)	[G]	AZ=(D.MS)
	or			FIELD Δ =?
8b	toward beginning point	R. AZ+ (D.MS)	[H]	AZ=(D.MS)
	or			FIELD Δ =?
	Input reference bearing toward beginning			
	point in place of azimuth			
8c	Input reference bearing and	BRG (D.MS)	[ENTER↑]	
	quadrant code	QD	[B]	AZ=(D.MS)
			[R/S]*	HD OR SD ↑Δ
	Ignore prompt and proceed			
9	Input field angle, either angle right	Δ (D.MS)	[G]	AZ=(D.MS)
				HD OR SD ↑Δ
	or angle left	Δ (D.MS)	[CHS] [G]	AZ=(D.MS)
				HD OR SD ↑Δ
	or deflection right	DEF (D.MS)	[H]	AZ=(D.MS)
				HD OR SD ↑Δ
	or deflection left	DEF (D.MS)	[CHS] [H]	AZ=(D.MS)
				HD OR SD ↑Δ
10a	Input horizontal distance or	HD	[D]	HD=
	go to step 10b		[R/S]*	N#=
			[R/S]*	E#=
			[R/S]*	NEXT COURSE?
	If display shows "RDY 01 OF 02"			RDY 01 of 02

*[R/S] required if printer is not used.

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	go to step 10c; otherwise go to step 7			
10b	Input slope distance and	SD	[ENTER↑]	
	vertical or zenith angle.	$\frac{\Delta}{(D.MS)}$	[I]	HD=
			[R/S]*	N#=
			[R/S]*	E#=
			[R/S]*	NEXT COURSE?
	If display shows - - -			or RDY 01 OF 02
	go to step 9c; otherwise go to step 10.			
10c	Either record data on side 1 of card			RDY 02 OF 02
	then on side 2 (write order no. on card)			
	then go to step 4, or			
	clear display and go to step 4.		[←] [R/S]	NEXT COURSE?
11	To delete last traverse course:			
	(error correction)		[J]	(previous) N#=
				(previous) E#=
				NEXT COURSE?
12	Input closing coordinates	Nc	[ENTER↑]	
		Ec	[E]	LAT ERR=
				DEP ERR=
				RDY 01 OF 02
	Either record data on side 1			RDY 02 OF 02
	and side 2 of card			0.0000
	or clear		[←] [R/S]	0.0000
13	To perform traverse adjustments using			
	compass rule go to step 14. For adjust-			
	ment using Crandall's rule go to step 19.			
14	Load compass rule program			

*[R/S] needed if printer is not used.

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
15	Execute program		[XEQ] AACMP	LAST CARD
				CARD
16	Read in side 1 and 2 of last card			
	recorded			N1=
			[R/S]*	E1=
			[R/S]*	AZ1=
			[R/S]*	HD1=
			[R/S]*	N2=
			[R/S]*	E2=
				.
				.
				.
17	Further data card input, if any, will be			
	prompted for by			CARD #
				CARD
	read in card			N#=
				E#=
				.
				.
				.
18	Repeat step 17 for all data cards			
19	Key in Crandall's rule program			
20	Execute program and go to step 16		[XEQ] AACDL	LAST CARD
				CARD

*[R/S] required if printer is not used.

Program Listings

Traverse for Auto Adjust Routines

```
 01◆LBL "TRV
AA"                  Initialize
 02 DEG
 03 SF 21
 04 SF 27
 05 CLRG
 06 CF 01
 07 CF 02
 08 SF 03
 09◆LBL 00           Initialize count-
 10 9                down
 11 STO 25
 12 FIX 4
 13 FS? 01
 14 GTO 09           Not 1st time
 15 SF 01            thru?
 16 "BEG N↑
BEG E"               1st time thru
 17 PROMPT
 18◆LBL A            Input Beg N,
 19 STO 23           Beg E
 20 STO 21
 21 RDN
 22 STO 22
 23 STO 20
 24 XEQ 05
 25 "COURSE
1"
 26 PROMPT
 27◆LBL G            RAZ-/Δ
 28 CF 00            1st time thru?
 29 FS?C 03
 30 SF 02            Reference azi-
 31 GTO 07           muth away
 32◆LBL B            Bearing, Quad-
 33 SF 00            rant input
 34◆LBL 07
 35 FS? 00           BRG input?
 36 GTO 01
 37 PI
 38 R-D
 39 HMS+
 40 GTO 08
 41◆LBL C            AZ input
 42 SF 00
 43◆LBL 08
 44 HR
 45 FS? 00           Bearing input?
 46 GTO 02
 47 RCL 24
 48 +
 49 GTO 02
 50◆LBL 01           BRG, QD→ AZ
 51 X<>Y
 52 HR
 53 X<>Y
 54 90
 55 *
 56 -
 57 LASTX
 58 X<>Y
 59 SIN
 60 ASIN
 61 ABS
 62 -
 63◆LBL 02           AZ
 64 1
 65 P-R
 66 R-P
 67 X<>Y
 68 X>0?
 69 GTO 03
 70 360
 71 +
 72◆LBL 03
 73 STO 24
 74 STO IND          Store AZ
25
 75 HMS
 76 ADV
 77 "AZ"
 78 FS? 02           Reference input?
 79 "⊢ REF"
 80 "⊢="
 81 ARCL X
 82 AVIEW
 83 "HD OR S
D ↑ ◿"
 84 FS?C 02          If ref input,
 85 "FIELD ◿         prompt for FLD◁
=?"                  else HD or S,◁
 86 PROMPT
 87◆LBL I            SD ↑ ◁
 88 HR
 89 X<>Y
 90 P-R
 91 X<=Y?            Convert SD & ◁
 92 X<>Y             tr HD
 93◆LBL D
 94 "HD="            Output HD
 95 ARCL X
 96 AVIEW
 97 10
```

Program Listings

```
 98 ST+ 25
 99 CLX
100 RDN
101 STO IND
25                    Store HD
102 ST+ 10
103 10
104 ST- 25
105 CLX
106 RDN
107 RCL IND
25                    AZ
108 X<>Y
109 P-R
110 RCL 22
111 +                 Σ+ (LAT)
112 STO 22
113 X<>Y
114 RCL 23
115 +
116 STO 23
117 XEQ 05            Output N, E
118 DSE 25            if n=0 record
119 GTO 09            reg's 0 - 25
120 .025              else
121 WDTAX             prompt for next
122 GTO 00            course
123◆LBL 09
124 "NEXT CO
URSE?"
125 PROMPT
126◆LBL H
127 CF 00
128 FS?C 03
129 SF 02
130 GTO 08
131◆LBL E             Nc ↑ Ec
132 X<>Y
133 RCL 22
134 -
135 STO 22
136 CHS
137 ADV
138 "LAT ERR          Output errors
="
139 ARCL X
140 AVIEW
141 X<>Y
142 RCL 23
143 -
144 STO 23
145 CHS
146 "DEP ERR
="
147 ARCL X
148 AVIEW
149 10
150 ST+ 25
151 CLX
152 RDN
153 ISG 25
154◆LBL 04
155 RCL IND           Make last HD neg-
25                    ative
156 CHS
157 STO IND
25
158 10
159 ST- 25
160 CLX
161 RDN
162 RCL 00
163 STO 24            # of points
164 CLD
165 .025
166 WDTAX             Prompt for rec-
167 CLX               ord of Reg 00-25
168 STOP
169◆LBL J             Delete
170 2                 decrement # of
171 ST- 00            points
172 ISG 25
173◆LBL 04
174 RCL IND           East AZ
25
175 10
176 ST+ 25
177 CLX
178 RDN
179 RCL IND           Last HD
25                    Σ- (HD)
180 ST- 10
181 10
182 ST- 25
183 CLX
184 RDN
185 CHS
186 P-R
187 RCL 22            Reverse Traverse
188 +
189 STO 22
190 X<>Y
191 RCL 23
```

Program Listings

```
192 +
193 STO 23
194 ISG 25
195*LBL 04
196 RCL IND
25                  Reset last AZ
197 STO 24
198 DSE 25
199 XEQ 05
200 GTO 09          Output N, E
201*LBL 05          Output N, E
202 1 E             routine
203 ST+ 00
204 FIX 0
205 CF 29
206 "N"
207 ARCL 00
208 "⊢="
209 FIX 4
210 ARCL 22
211 AVIEW
212 FIX 0
213 "E"
214 ARCL 00
215 "⊢="
216 FIX 4
217 SF 29
218 ARCL 23
219 AVIEW
220 RTN
221 .END.
```

Auto Adjust for Compass Rule

```
 01*LBL "AAC
MP"
 02 DEG
 03 CF 21           Don't stop on
 04 "LAST CA        AVIEW
RD"
 05 BEEP
 06 AVIEW           Prompt for last
 07 PSE             card
 08 SF 21           Stop on AVIEW
 09 CLD
 10 .025
 11 RDTAX           Read regs 00-25
 12 RCL 22
 13 RCL 10
 14 /
 15 STO 22
 16 RCL 23
 17 RCL 10
 18 /
 19 STO 23
 20 9
 21 RCL 24
 22 X<=Y?
 23 SF 02           No more data
 24 0               required
 25 STO 24
 26 STO 00
 27 XEQ 12
 28 FS?C 02
 29 GTO 02
 30*LBL 00
 31 19
 32 STO 25
 33 RCL 00
 34 1               Increment card
 35 +               no
 36 FIX 0
 37 CF 21
 38 "READ CA
RD "
 39 CF 29
 40 ARCL X
 41 SF 29
 42 FIX 4
 43 BEEP
 44 AVIEW           Prompt for proper
 45 PSE             card
 46 SF 21
 47 CLD
 48 .019
 49 RDTAX           Read regs 00-19
 50 RDN
 51 STO 00
 52*LBL 02
 53 9               Set (i) address
 54 STO 25
 55*LBL 03
 56 RCL IND
25                  (AZ)
 57 10
 58 ST+ 25
 59 CLX
 60 RDN
 61 RCL IND         HD
25
 62 X<0?
 63 SF 02           Last course?
 64 ABS
 65 STO IND
25
```

Program Listings

```
 66 P-R
 67 RCL 22          Adjust LAT
 68 RCL IND
25
 69 *
 70 +
 71 X<>Y
 72 RCL 23          Adjust DEP
 73 RCL IND
25
 74 *
 75 +
 76 10
 77 ST- 25
 78 CLX
 79 RDN
 80 X<>Y
 81 R-P             Adjust AZ & HD
 82 X<>Y
 83 X>0?
 84 GTO 04
 85 360             Add 360 if AZ is
 86 +               negative
 87◆LBL 04
 88 HMS
 89 "AZ"
 90 XEQ 11          Output AZ
 91 HR
 92 X<>Y
 93 "HD"
 94 XEQ 11
 95 P-R             Output HD
 96 RCL 20
 97 +               Σ(N)
 98 STO 20
 99 X<>Y
100 RCL 21          Σ(E)
101 +
102 STO 21
103 XEQ 12
104 FS?C 02
105 STOP            Halt if finished
106 DSE 25          Loop for next
107 GTO 03          course
108 GTO 00          or next card
109◆LBL 12          Output routine
110 RCL 24
111 1
112 +
113 STO 24
114 ADV
115 RCL 20
116 "N"
117 XEQ 11          Output N
118 RCL 21
119 "E"
120◆LBL 11          and output E
121 FIX 0
122 CF 29
123 ARCL 24
124 "⊢="
125 FIX 4
126 ARCL X
127 SF 29
128 AVIEW
129 CLD
130 RTN
131 END
```

Auto Adjust for Crandall's Rule

```
 01◆LBL "AAC
DL"
 02 DEG
 03 CF 21
 04 "LAST CA
RD"
 05 BEEP
 06 AVIEW
 07 PSE             Prompt for last
 08 SF 21           card
 09 CLD
 10 .025
 11 RDTAX
 12 RCL 23          Read in regs
 13 CHS             00-25
 14 STO 21
 15 RCL 22
 16 CHS
 17 STO 20
 18 9
 19 RCL 24
 20 0
 21 STO 22          Clear Σ's
 22 STO 23
 23 STO 24
 24 STO 00
 25 RDN
 26 X<=Y?
 27 GTO 01          Data needed?
 28◆LBL 00
 29 RCL 00
 30 1               Increment card #
 31 +
 32 .019            Reg 00-19
```

Program Listings

```
33 XEQ 12
34◆LBL 01
35 9
36 STO 25
37◆LBL 02
38 XEQ 07
39 XEQ 08
40 RCL 22
41 +
42 STO 22
43 RDN
44 X<>Y
45 XEQ 08
46 RCL 23
47 +
48 STO 23
49 RDN
50 *
51 XEQ 09
52 RCL 24
53 +
54 STO 24
55 XEQ 14
56 FS?C 02
57 GTO 03
58 DSE 25
59 GTO 02
60 GTO 00
61◆LBL 03
62 RCL 24
63 RCL 21
64 *
65 RCL 20
66 RCL 23
67 *
68 -
69 RCL 23
70 RCL 22
71 *
72 RCL 24
73 X↑2
74 -
75 /
76 STO 00
77 RCL 20
78 RCL 24
79 *
80 RCL 21
81 RCL 22
82 *
83 -
84 RCL 22
```

Data card input

Σ's

$\Sigma \frac{L^2}{(HD)}$

$\Sigma \frac{D^2}{(HD)}$

$\Sigma \frac{(L)(D)}{(HD)}$

Last course?

Loop for next course

Loop for next card

Compute A & B

A

```
85 RCL 23
86 *
87 RCL 24
88 X↑2
89 -
90 /
91 STO 23
92 RCL 00
93 STO 22
94 1
95 ENTER↑
96 .021
97 XEQ 12
98 1
99 STO 24
100 XEQ 13
101◆LBL 05
102 XEQ 07
103 RCL 22
104 *
105 X<>Y
106 RCL 23
107 *
108 +
109 XEQ 14
110 XEQ 07
111 RDN
112 X<>Y
113 *
114 LASTX
115 R↑
116 *
117 XEQ 09
118 X<>Y
119 XEQ 09
120 XEQ 14
121 XEQ 07
122 RDN
123 +
124 X<>Y
125 R↑
126 +
127 XEQ 14
128 R-P
129 X<>Y
130 X>0?
131 GTO 06
132 360
133 +
134◆LBL 06
135 HMS
136 "AZ"
137 XEQ 11
```

B

A

1st card

Reg 00-21

Reinput 1st card

Init. point

Output BEG N, E

Adjustment

[(A)(L)+(B)(D)]
(D)

[(A)(L)+(B)(D)]
(L)

l

d

Output AZ

Program Listings

```
138 HR
139 X<>Y
140 "HD"
141 XEQ 11
142 P-R
143 RCL 20
144 +
145 STO 20
146 RDN
147 RCL 21
148 +
149 STO 21
150 1
151 RCL 24
152 +
153 STO 24
154 XEQ 13
155 FS?C 02
156 STOP
157 DSE 25
158 GTO 05
159 RCL 00
160 1
161 +
162 .019
163 XEQ 12
164 GTO 05
165◆LBL 07
166 RCL IND
25
167 10
168 ST+ 25
169 CLX
170 RDN
171 RCL IND
25
172 X<0?
173 SF 02
174 ABS
175 P-R
176 RTN
177◆LBL 08
178 ENTER↑
179 X↑2
180◆LBL 09
181 RCL IND
25
182 ABS
183 ⁄
184 RTN
185◆LBL 12
186 FIX 0
187 CF 21
```

Output adj HD

Σ(N)

Σ(E)

Inc. point #

Output adj N, E
Last course

Loop for next course

Data card input

AZ

HD

Last course

L, D

HD

Data card read routine

Don't stop on AVIEW

```
188 "READ CA
RD "
189 CF 29
190 ARCL Y
191 SF 29
192 FIX 4
193 BEEP
194 AVIEW
195 PSE
196 SF 21
197 CLD
198 RDTAX
199 RDN
200 STO 00
201 9
202 STO 25
203 RTN
204◆LBL 13
205 ADV
206 "N"
207 RCL 20
208 XEQ 11
209 "E"
210 RCL 21
211◆LBL 11
212 FIX 0
213 CF 29
214 ARCL 24
215 "⊢="
216 FIX 4
217 ARCL X
218 SF 29
219 AVIEW
220 CLD
221 RTN
222◆LBL 14
223 10
224 ST- 25
225 CLX
226 RDN
227 RTN
228 .END.
```

No decimal point

Decimal point

Prompt for proper card

Stop on AVIEW
Read card

Store card #

Indirect address

Output routine

Change i address by 10

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Print #/Card #	50	
	↑		
05	Azimuths	55	
	↓		
10	Σ+ (HD)	60	
	↑		
15	Distances	65	
	↓		
20	Used		
	Used		
	Used		
	Used		
	Used		
25	Used		
30			
35			
40			
45			

KEYBOARD OVERLAY



STATUS

SIZE 026 TOT. REG. * USER MODE

ENG ____ FIX 4 SCI ____ ON ____ OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		TRAVERSE PROGRAM	
0		Bearing input	Ref AZ input
1	S	Prompt for next course	Prompt for BEG N,E
2	C	Prompt for FIELD◁	Prompt for HD or SD
3	S	Input ref AZ	Input AZ
21	S	Halt on AVIEW	PSE on AVIEW
27	S	User mode	
29		Dec pt in FIX 0	No dp in FIX 0
		COMPASS/CRANDALL'S RULE PROGRAM	
2	C	More data to read	No more data
21		Halt on AVIEW	PSE on AVIEW
29		No decimal point	Decimal point

ASSIGNMENTS

TRVAA

FUNCTION	KEY	FUNCTION	KEY
Input BEG N,E	A	REF AZ-/◁	G
Input BRG, QD	B	REF A,Z+/DEF	H
Input AZ	C	SD ↑ ◁	I
Input HD	D	error correct	J
Input Nc, Ec	E		

* 89 registers for Traverse for Auto Adjust Routines
61 registers for Auto Adjust for Compass Rule
83 registers for Auto Adjust for Crandall's Rule

TRAVERSE FOR AUTO ADJUST ROUTINES
PROGRAM REGISTERS NEEDED: 63

HEWLETT PACKARD SOLUTION BOOK: SURVEYING

ROW 1 (1 - 4)

ROW 2 (4 - 12)

ROW 3 (12 - 16)

ROW 4 (16 - 20)

ROW 5 (20 - 25)

ROW 6 (25 - 30)

ROW 7 (30 - 37)

ROW 8 (38 - 46)

ROW 9 (46 - 55)

ROW 10 (56 - 68)

ROW 11 (69 - 76)

ROW 12 (77 - 80)

ROW 13 (80 - 83)

ROW 14 (83 - 85)

ROW 15 (85 - 93)

ROW 16 (94 - 100)

ROW 17 (101 - 108)

ROW 18 (109 - 117)

ROW 19 (117 - 122)

ROW 20 (122 - 124)

ROW 21 (124 - 130)

ROW 22 (131 - 138)

ROW 23 (138 - 143)

ROW 24 (144 - 147)

ROW 25 (147 - 155)

ROW 26 (156 - 164)

ROW 27 (165 - 171)

ROW 28 (172 - 179)

ROW 29 (180 - 188)

ROW 30 (189 - 196)

ROW 31 (197 - 203)

ROW 32 (203 - 209)

ROW 33 (209 - 215)

ROW 34 (216 - 221)

AUTO ADJUST FOR COMPASS RULE

PROGRAM REGISTERS NEEDED: 35

HEWLETT PACKARD
SOLUTION BOOK:
SURVEYING

ROW 1 (1 - 4)

ROW 2 (4 - 8)

ROW 3 (8 - 15)

ROW 4 (15 - 23)

ROW 5 (24 - 31)

ROW 6 (31 - 38)

ROW 7 (38 - 41)

ROW 8 (41 - 48)

ROW 9 (49 - 57)

ROW 10 (58 - 66)

ROW 11 (67 - 75)

ROW 12 (76 - 85)

ROW 13 (85 - 92)

ROW 14 (93 - 99)

ROW 15 (100 - 106)

ROW 16 (107 - 115)

ROW 17 (115 - 121)

ROW 18 (122 - 127)

AUTO ADJUST FOR COMPASS RULE

ROW 19 (128 - 131)

AUTO ADJUST FOR CRANDALL'S RULE
PROGRAM REGISTERS NEEDED: 58

HEWLETT PACKARD SOLUTION BOOK: SURVEYING

ROW 1 (1 - 4)

ROW 2 (4 - 8)

ROW 3 (8 - 14)

ROW 4 (15 - 22)

ROW 5 (23 - 32)

ROW 6 (32 - 38)

ROW 7 (39 - 45)

ROW 8 (46 - 53)

ROW 9 (54 - 59)

ROW 10 (60 - 67)

ROW 11 (68 - 77)

ROW 12 (77 - 85)

ROW 13 (85 - 94)

ROW 14 (95 - 100)

ROW 15 (100 - 108)

ROW 16 (109 - 117)

ROW 17 (117 - 122)

ROW 18 (123 - 132)

AUTO ADJUST FOR CRANDALL'S RULE

HEWLETT PACKARD SOLUTION BOOK: SURVEYING

ROW 19 (132 - 139)

ROW 20 (140 - 146)

ROW 21 (147 - 154)

ROW 22 (154 - 162)

ROW 23 (162 - 168)

ROW 24 (168 - 178)

ROW 25 (179 - 188)

ROW 26 (188 - 190)

ROW 27 (190 - 198)

ROW 28 (199 - 208)

ROW 29 (208 - 214)

ROW 30 (215 - 222)

ROW 31 (223 - 228)

NOTES

NOTES

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

SURVEYING

SPIRAL CURVE LAYOUT
TWO INSTRUMENT RADIAL SURVEY
EDM SLOPE REDUCTION
STADIA REDUCTION
THREE WIRE LEVELING
AZIMUTH OF THE SUN
TAPING REDUCTION
TRIANGLE SOLUTIONS
TRAVERSE FOR AUTO ADJUST ROUTINES, AUTO ADJUST FOR COMPASS RULE, AUTO ADJUST FOR CRANDALL'S RULE

00041-90141 Rev. C 5/83 Printed in U.S.A.